إنشاء الفئات والكائنات الموجهة في الاكسس
بقلم /عارف حسان   الرابط
مقدمة:
فهذا -حسب ظني- أول بحث باللغة العربية يشرح الفئات والكائنات الموجهة في الفيجوال بيسيك للتطبيقات شرحاً لا أقول عنه كاملاً -فالكمال لله وحده سبحانه وتعالى- بل كافياً ليضع قدمك على أرضية صلبة تستطيع أن تنطلق منها إلى فضاءات برمجية أكثر رحابة. وبالرغم من أنني سأحاول أن أشرح فيه جميع الأساسيات, والمفاهيم, والمصطلحات, مع بعض الأمثلة التي يمكن أن تواجه المبرمج على أرض الواقع, وليس أمثلة لمجرد شرح الفكرة فقط, إلا أن هناك بعض المهارات البرمجية لا يمكن اكتسابها من مجرد القراءة فقط, بل يجب عليك التطبيق والتجربة لاكتسابها, فرب حامل فقه ليس بفقيه, فمحدثكم ليس فقيهاً في البرمجة, بل هو حامل فقه لمن هو أفقه منه, فلعلكم بعد قراءتكم لهذا البحث تدهشونني بأفكار لم تكن يوما لتخطر لي على بال.
وقد قررت بعد طول تفكير أن يكون هذا البحث على شكل مواضيع يتم نشرها في قسم الوحدات النمطية والماكرو ثم يتم عمل رابط لها في هذا البحث على شكل عناوين, بحيث –عند الانتهاء من البحث- يسهل على القارئ مشاهدة جميع عناوين البحث, وبمجرد النقر على أي عنوان سيتم نقله مباشرة إلى ذلك الموضوع.

لمن هذا البحث:
برمجة الفئات من الموضوعات المتقدمة في البرمجة, ولكي تستطيع أن تفهم هذا البحث جيداً فيفترض أن تكون لديك خلفية جيدة أو متوسطة في برمجة الـ vba , كما يفضل -أيضاً- أن تكون لك خبرة في تصميم قواعد البيانات.

الموضوع الأول وتجد فيه ما يلي:
مدخل لتعريف الفئات والكائنات

1- الوحدات النمطية القياسية
2- الوحدات النمطية للفئة
[=1- نمطية الفئة المترافقة مع النماذج والتقارير
[=]2- نمطية الفئة غير المترافقة مع أي نموذج أو تقرير

3. ماهي الكائنات؟

4. ما هي أعضاء الفئة؟

5. ما الفرق بين الوحدات النمطية للفئة والوحدات النمطية القياسية؟.

1- فائدة البرمجة باستخدام الوحدات النمطية للفئات
2- متى يلجأ المبرمج إلى برمجة الفئات؟!!
3- علاقة الفئات بالبرمجة الموجهة للكائنات:


الموضوع الثالث: وتجدون فيه ما يلي:
مبادئ وأساسيات ومفاهيم البرمجة الموجهة للكائنات OOP

المبدأ الأول: الفئات Classes .
المبدأ الثاني: الكائنات Objects .
المبدأ الثالث: التغليف Encapsulation .
المبدأ الرابع: الوراثة Inheritance .
المبدأ الخامس: تعدد الأشكال أو تعدد الأوجه Polymorphism .


الموضوع الرابع وتجدون فيه ما يلي:
المبدأ الرابع: الوراثة Inheritance .

وراثة الواجهات Interface inheritance :
ماهي الواجهات Interfaces ؟ :
مثال عن الواجهات:
ماهي الفئات المجردة Abstract Classes أو فئات الواجهة Interface Classes :
الموضوع الخامس وتجدون فيه ما يلي:
المبدأ الخامس: تعدد الأشكال Polymorphism .
تعدد الأشكال أو تعدد الأوجه Polymorphism مع مثال يشرحه
الموضوع السادس وتجدون فيه ما يلي:
تقنيات بناء الفئات وإنشاءها
أسس بناء الفئات:
إنشاء كائن الفئة :
التصريح عن المتغيرات الداخلية للفئة:

الموضوع السابع وتجدون فيه ما يلي:
إنشاء خصائص الفئات
بين خصائص الفئة Class Properties وسمات الكائنات Object attributes :
إنشاء خصائص الفئة:
ماهي إجراءات الخاصية Property Procedures ولماذا نستخدمها؟:


الموضوع الثامن وتجدون فيه ما يلي:
تمثيل الفئات Classes Instantiating
إنشاء كائن مثيل للفئة Instance :
الكلمتان المفتاحيتان Set و New :
اختصار سطر في الكود قد يكون مكلفاً:

الموضوع التاسع وتجدون فيه ما يلي:
إنشاء الطرق Methods في الفئات
إنشاء الطرق Creating methods :
متى نستخدم إجراءات الخاصية, ومتى نستخدم إجراءات الدالة؟:
تطبيق عملي: 

الموضوع العاشر وتجدون فيه ما يلي:
أحداث الفئات المبنية
الحدثان Initialize و Terminate :
فيم يستخدم الحدثان؟:
تطبيق عملي:
الموضوع الحادي عشر وتجدون فيه ما يلي:
التسلسل الهرمي للفئات Class Hierarchies
ما هي الفئات الهرمية:
كيفية تمثيل العلاقات بين الفئات:
الوراثة بالتفويض Delegation :
تطبيق عملي:
اختبار الفئة المشتقة:


الموضوع الثاني عشر وتجدون فيه ما يلي:
فئات المجموعة Collection Classes - الجزء الأول

كائن المجموعة Collection object سفينة نوح في البرمجة:
الفرق بين المجموعات المبنية في الاكسس و المجموعة التي ننشئها بالكود:


الموضوع الثالث عشر وتجدون فيه ما يلي:
فئات المجموعة Collection Classes الجزء الثاني
إنشاء فئة مجموعة Collection Class :
المشكلة مع فئات المجموعة:
طرق حل المشكلة:

[=2]1- التقنية الأولى إضافة دالة AllItems إلى الفئة:
[=2] 2- التقنية الثانية تصدير وإعادة استيراد الفئة:


الموضوع الرابع عشر وتجدون فيه ما يلي:
استخدام الفئات Events & Classes والأحداث
لماذا نحتاج إلى إنشاء الأحداث في الفئات:
كيفية إنشاء الأحداث في الفئات:
الكلمة المفتاحية WithEvents :
مثال عملي: 

الموضوع الخامس عشر وتجدون فيه ما يلي:
الفئات المجردة والفئات المنفذة Abstract & Implementation Classes

ما هي الفئات المجردة:
ما حاجتنا إلى الفئات المجردة؟:
الكلمة المفتاحية Implements :
مثال عملي:
إنشاء الفئة المنفذة:

الموضوع السادس عشر وتجدون فيه ما يلي:
الفئة المجردة الهجينة Hybrid Abstract Class
ماهي الفئة المجردة الهجينة:
كيفية إنشاء فئة هجينة:

الموضوع السابع عشر وتجدون فيه ما يلي:
معالجة الأخطاء في الفئات
تصيد الأخطاء Error Trapping :
كيف يتعامل الفيجوال بيسيك للتطبيقات مع الأخطاء:
كيفية معالجة الأخطاء في الفئات:
التحكم في نصوص رسائل خطأ وقت التشغيل: 
إظهار رسائل أخطاء معرفة من قبل المبرمج Raising Custom_Defined Errors :
تمرير الأخطاء في الوحدات النمطية للفئات:

الموضوع الثامن عشر وتجدون فيه ما يلي:
عمر و مجال متغيرات الفئة ومثيلاتها

مجال وفترة حياة متغيرات الفئة:
[=2]المتغيرات على مستوى الفئة:
المتغيرات على مستوى إجراءات الفئة:
متغيرات الفئة التي تشير إلى كائنات أخرى:
[=3]حالات المرجع الدائري Circular Reference Issues :
[=4]الإنهاء المتأخر Delayed Termination :
الكائنات اليتيمة Orphaned Objects :
الإنهاء السليم Proper Termination :


الموضوع التاسع عشر وتجدون فيه ما يلي:
الفئات والنماذج
فتح نسخ متعددة من نموذج:
كيفية إنشاء وحدة نمطية لنموذج:
كيفية الإشارة بالكود إلى نسخ متعددة مفتوحة من نموذج:
الفئات والنماذج المرتبطة:
ربط نموذج إلى مجموعة سجلات بتقنية الـ DAO :

مسك الختام:
كيف تشير الفئة إلى نفسها؟:
فحص كائن أو تركيب Structure بالدالة TypeOf :
كيف تنشئ العلاقات بين الفئات؟:
ما هي الدالة ObjPtr وما أصلها؟:
تطبيق عملي

مراجع هذا البحث:
المراجع العربية:
1-.فيجوال بيسيك للجميع تركي العسيري
2- التجوال في كتاب Beginning Access 2003 VBA ... أبو يوسف1
3- موسوعة ويكيبديا العربية
المراجع الأجنبية:
1- VBA Devloper's Handbook Second Edition - Ken Getz, Mike Gilbert
2- Access 2007 VBA Programmer's Reference - Teresa Hennig & Others
3- Access 2010 VBA Programming INSIDE OUT - Andrew Couch
4- Microsoft® Access VBA Programming for the Absolute Beginner - Michael Vine
5- Beginning Access 2003 VBA - Denise Gosnell

مدخل لتعريف الفئات والكائنات
هذا هو الموضوع الأول في سلسلة البحث المسمى: إنشاء الفئات والكائنات الموجهة في الآكسس
كما تعلمون فهناك نوعان من الوحدات النمطية في الآكسس:
1- الوحدات النمطية القياسية: والتي -لاشك- أنكم قد تعاملتم معها مسبقاً, فهي التي تستخدم للبرمجة الإجرائية, وهي لا تدخل في موضوعنا هذا, اللهم إلا إذا أردنا أن نختبر بها الفئات.
2- الوحدات النمطية للفئات: وهي نوعان أيضاً:
1- وحدات نمطية للفئة مترافقة مع النماذج والتقارير, فالنماذج عبارة عن فئة, والتقارير كذلك.
2- وحدات نمطية للفئة غير مترافقة مع أي نموذج أو تقرير, بل يتم إنشاؤها عن طريق المستخدم (المقصود بالمستخدم هو مستخدم برنامج الاكسس الذي هو أنت كمبرمج, وليس العميل النهائي الذي يستخدم برنامجك).
يمكن تشبيه الفئات بالقالب الذي يتم على أساسه عمل عدة نسخ, كالقالب في الوورد (يمكن عمل عدة نسخ على أساسه, وتستطيع أن تملأ كل نسخة مأخوذة من القالب ببيانات مختلفة عن النسخة الأخرى), ويمكن تشبيه الفئات أيضاً بالمخطط المعماري, حيث يمكن بناء عدة أبنية متشابهة وفقاً لمخطط واحد فقط, وكما يضع المهندسون في مخططات البناء رسوماً توضح مداخل البناء, ومخارجه, وأماكن الجلوس, والنوم, والطعام, وغيره, كذلك يضع المبرمج في الفئة أكواداً توضح خصائص الفئة, ووظائفها, وأحداثها, وما الذي ينبغي عليها عمله عند استدعائها, أو عند انتهائها من تنفيذ عملها.

ماهي الكائنات:
من أجل أن تعرف ما هي الفئات, يجب أن تعرف أولاً ماهي الكائنات, لأن الفئات عبارة عن كائنات. فكل فئة من الفئات تعتبر كائناً خاصاً مميزاً له خصائص تميزه, ووظائف يقوم بها, وأحداثٌ تجري عليه. هذه الخصائص والوظائف والأحداث تسمى أعضاء الفئة. ولكي تصبح الصورة أقرب إلى ذهنك سأعطيك أمثلة من الواقع لكي تعرف مفهوم الكائنات وبالتالي مدخلاً إلى فهم الفئات.
جميع ما يمكن أن يقع عليه ناظريك يعتبر كائناً, فالغرفة التي تجلس فيها الآن كائن, وكل شئ موجود بداخلها يعتبر كائناً .. آهـ .. قف هنا .. هذا يعني أن الكائن الواحد يمكن أن يحوي بداخله عدة كائنات .. فهل هذا صحيح؟!! .. نعم هذا صحيح 100% .. يمكن للكائن الواحد أن يضم بداخله كائنات أخرى, وانظر إلى حاسوبك الذي بين يديك, فهو أيضاً كائن, وبنفس الوقت فهو يحتوي على عدة كائنات, فالشاشة كائن لها صفات الطول والعرض, والنوع, والوظيفة, ولوحة المفاتيح كائن, لها صفات ووظائف, والفأرة كائن, وهكذا .. لكن حاسوبك ليس بدعاً من الحواسيب, بل لقد تم صنع عدة الآف أو أكثر من نفس الفئة من الحواسيب, وبنفس الموديل طبعاً مع اختلاف الرقم التسلسلي لكل جهاز .. آهـ .. قف هنا قليلاً لو سمحت ..هذا يعني أنه يمكن إنشاء عدة كائنات اعتماداً على فئة واحدة .. أليس كذلك؟!! .. نعم ذلك صحيح أيضاً فيمكن لنا أن ننشئ من الفئة الواحدة عدة كائنات متشابهة في الخصائص والصفات, لكنها قد تحمل قيماً مختلفة لهذه الصفات .. كيف يكون ذلك؟!!
قلنا في المثال السابق أن جميع الحواسيب التي تم صنعها لفئة معينة لها نفس المواصفات, لكن كلاً منها يختلف في خاصية الرقم التسلسلي, ثم أن لديها جميعاً خاصية القرص الصلب, لكنك لا تدري أي نوع من البيانات والمعلومات سيحويها بعد عملية الشراء أليس كذلك؟!!
وإليك مثال آخر: الإنسان كائن .. وفئة الإنسان تشترك في صفات عامة منها الاسم, والطول, والوزن, كما أن لها طرقاً كالتفكير, والكلام, والحركة.. وتحتوي فئة الإنسان على كائنات منها عارف حسان وطوله 167سم ووزنه 75كغ, ومنها أنت واسمك كذا وطولك كذا ووزنك كذا .. صحيح أننا كائنات نتبع نفس الفئة ولنا نفس الخصائص, لكننا نختلف في قيمة هذه الخصائص من شخص لآخر.

ماهي أعضاء الفئة؟
تحتوي الفئات غالباً على كل أو بعض مما يلي:
1- خصائص (Properties) : ليس شرطاً أن تحتوي الفئات على خصائص, فقد تجد فئات بها إجراءات دالة فقط, أو روتينات فرعية فقط, أو تكون عبارة عن فئات واجهة (فئات مجردة) ليس فيها أكواد تنفيذية. (سيتم شرحها لاحقاً).
2- طرق (Methods) : أيضاً ليس شرطاً أن تحتوي الفئات على الطرق, (الطرق: هي الدالات, والروتينات الفرعية).
3- أحداث (Events) : قد تحتوي الفئة على الأحداث, وقد تخلو من أي حدث.
فهذه الخصائص, والطرق, والأحداث, تسمى أعضاء الفئة.

ما الفرق بين الوحدات النمطية للفئة والوحدات النمطية القياسية؟.
الاختلاف البارز بينهما هو أنه إذا أردنا استخدام الفئة فلابد لنا من عملية تسمى تمثيل أو استنساخ الفئة Instantiating (سيتم شرح هذا العملية لاحقاً), بينما في الوحدات النمطية القياسية فلا نحتاج إلى هذه العملية.
إلى هنا ينتهي هذا الموضوع, وستعرف في الموضوع القادم -إن شاء الله- فائدة وأهمية الفئات, ومتى يلجأ المبرمجون إليها, وما علاقة الفئات بمصطلح أو بنمط البرمجة الموجهة للكائنات.
أهمية الفئات والبرمجة الكائنية
هذا هو الموضوع الثاني في سلسلة البحث المسمى: إنشاء الفئات والكائنات الموجهة في الآكسس
عرفنا في الموضوع السابق أن الوحدات النمطية للفئة ما هي إلا قوالب يتم بموجبها إنشاء كائنات متشابهة في صفاتها وخصائصها العامة وسلوكها بشكل عام, لكنها قد تختلف في قيمة هذه الخصائص أو القيم العائدة منها, وضربنا لكم الأمثال. والآن سنتعرف على الفائدة المرجوة من هذا النمط من البرمجة, ومتى يقرر المبرمج أن يلجأ إليه.
فائدة البرمجة باستخدام الوحدات النمطية للفئات:
برمجة الفئات تمنح المبرمجين إمكانية تغليف وتنظيم ما يكتبون من أكواد وجعلها في وحدات برمجية سهلة الصيانة, وقابلة للتطوير مستقبلاً بكل سهولة, كما يمكن إعادة استخدامها من قبل المبرمج نفسه, أو من قبل مبرمجين آخرين بغض النظر عن اختلاف البرنامج أو المبرمج, حيث من المفترض أن تكون واجهات الفئات مستقلة ولها وظائف محددة تؤديها باستقلالية عن بقية مكونات البرنامج. كذلك فقد تكون برمجة الفئات من الوسائل الفعالة جداً في إنقاص كتلة الكود البرمجي إلى حد بعيد, وسأضرب لكم مثلاً في هذا المثال الرائع للأستاذ الملهم/ ناصر المصاروة والذي لا زلت أدرسه وأتعلم منه إلى اليوم. انقر هنا لمشاهدة الموضوع. 
في المثال السابق تجدون أن هناك حوالي 50 أو 49 صورة في النموذج المسمى Form1 ولكل صورة أربعة أحداث, فعند استخدام البرمجة الكائنية سوف نختصر الكثير من كتلة الكود المكتوب في النموذج, وسنوفر الجهد والوقت .. قارن الكود الموجود في النموذج Form1 مع الكود الموجود في المثال التالي ولاحظ الفرق في كتلة الكود.

متى يلجأ المبرمج إلى برمجة الفئات؟!!
يمكن لكثير من الأمور التي يمكن إنجازها عبر الفئات أن نقوم بإنجازها عبر دالة أو روتين فرعي نضعه في وحدة نمطية قياسية. لذلك فخذها قاعدة عامة: "ما يمكن إنجازه عبر دالة أو روتين فرعي بسيط بدون برمجة الفئات, فأنجزه بدونها" .. ولا تلجأ إلى برمجة الفئات في كل الأمور, بل اعرف متى يلجأ المبرمجون إلى برمجة الفئات, وستجد أنهم يلجأون إليها لأسباب منها:
1- تبسيط الكود البرمجى, وتقليل الأخطاء البرمجية.
2- يصبح حل بعض الصعوبات البرمجية أمراً ميسوراً عبر برمجة الفئات.
3- قد يقوم مبرمج الفئات بعمل برامج تكون مساعدة للمبرمجين الآخرين في عملية البرمجة, بحيث تكون هذه البرامج شبيهة بالمراجع أو المكتبات التي يمكن استدعاؤها عند الحاجة إليها. (مثل: برنامج صلاحيات المستخدمين للأستاذ/ مهند عبادي).
علاقة الفئات بالبرمجة الموجهة للكائنات:
مما سبق يتضح جلياً أن كل الحديث السابق يدور حول الكائنات, لنمهد الطريق أمامك لإنشاء كائنات خاصة بك. ويدعى هذا النمط من البرمجة بالبرمجة الموجهة للكائنات, أو البرمجة الكائنية, حيث تعتبر الفئات فيها وحدة البناء الرئيسية, والحجر الأساسي في بناء الكائنات, إذ يمكن لنا عبر الفئة أن ننتج عدداً غير محدود من الكائنات الخاصة بتلك الفئة .. قف هنا لو سمحت .. هل يمكن فعلاً إنتاج عدد غير محدود من الكا ئنات للفئة الواحدة؟!!, أم أن هذا زلة قلم؟!! .. الجواب: نعم .. يمكن إنشاء عدد غير محدود من الكائنات لأي فئة كانت, ويعود السبب إلى أن الكائنات لها آجال كآجالكم, وأعمار كأعماركم, ولها لحظة تولد فيها وأخرى تنجز فيها ما قدر لها مبرمجها إنجازه, ثم تموت ولا تعود إلى يوم الدين, ولأن عمر الكائنات -أخي الكريم- قد يكون قصيراً جداً, حيث يمكن خلال ثوان قليلة فقط ولادة وموت الآف الكائنات, وفي تلك الثواني القليلة بين لحظة ولادتها وموتها -وربما قبل أن يرتد إليك طرفك- تكون هذه الآلآف المؤلفة من الكائنات قد أنجزت ما هو مطلوب منها إنجازه, فمنها ما يقضى نحبه أثناء التشغيل, ومنها ما ينتظر وما بدلوا تبديلاً.
إذن فالحديث عن برمجة الفئات ماهو إلا حديث عن البرمجة الموجهة للكائنات Object Oriented Programming أو تسمى اختصاراً OOP وهذا النمط من البرمجة معروف بشكل أفضل في بعض لغات البرمجة مثل الجافا والسي بلس بلس أوالفيجوال دوت نت, وبشكل أقل في لغة الفيجوال بيسيك6 و الفيجوال بيسيك للتطبيقات, والسبب يعود إلى أن هاتين الأخيرتين لا تتوفر فيهما كامل إمكانيات البرمجة الموجهة للكائنات, وهذا يقودنا إلى عنوان الموضوع القادم -إن شاء الله- وهو مبادئ وأساسيات ومفاهيم البرمجة الموجهة للكائنات OOP إلى جانب عناوين أخر.
ملاحظة: 
جميع ما سبق وما يلي من المواضيع عبارة عن كتابة ارتجالية من الذاكرة مما سبق لي قراءته من المراجع الآنفة الذكر, لذلك فاحتمال الخطأ أو النسيان وارد الحدوث, فأرجو ممن يجد خطأ أو نقصاً أن يشركنا معه في معرفته, وسأكون له من الشاكرين.

مبادئ وأساسيات ومفاهيم البرمجة الموجهة للكائنات OOP
هذا هو الموضوع الثالث في سلسلة البحث المسمى: إنشاء الفئات والبرمجة الموجهة للكائنات في الآكسس
عرفنا في الموضوع السابق أهمية البرمجة باستخدام الفئات, وعرفنا أن هذا النمط من البرمجة تدعى بالبرمجة الموجهة للكائنات, وعرفنا أنه يمكن لنا أن نوفرعبرها الكثير من الوقت والجهد إما عن طريق اختصار أسطر الكود المكتوب, أو عن طريق إنشاء كائنات قابلة لإعادة استخدامها في عدة برامج, كما يتم فيها تبسيط كتابة الكود عبر تقسيم البرنامج إلى وحدات صغيرة تسمى كائنات - بدلاً من كائن واحد كبير - وهذا ما يجعل صيانة وتطوير البرامج أكثر سهولة. واليوم سنتعرف على مبادئ ومفاهيم هذه البرمجة, وبعضاً من مصطلحاتها الشائعة.
تختص نمط البرمجة الموجهة للكائنات بمبادئ ومفاهيم ومصطلحات خاصة بها, وتدعم لغة الفيجوال بيسيك للتطبيقات بعض هذه المبادئ بشكل كامل, وتدعم بعضها بشكل جزئي, بينما بعض المصطلحات ليست مدعومة في لغة الفيجوال بيسيك للتطبيقات على الإطلاق.
مبادئ البرمجة الموجهة للكائنات:
تحتوي البرمجة الموجهة للكائنات على خمسة مبادئ أو مفاهيم أساسية وهي:
المبدأ الأول: الفئات.
المبدأ الثاني: الكائنات.
المبدأ الثالث: التغليف.
المبدأ الرابع: الوراثة.
المبدأ الخامس: تعدد الأشكال أو تعدد الأوجه.
أما المصطلحات المستخدمة في هذا النمط من البرمجة فهي كثيرة, وسأذكر منها فقط ما يستخدم في لغة الفيجوال بيسيك للتطبيقات, وما يحضرني الآن, وهي: الفئة الأم, الفئة المشتقة, الفئات الهرمية, فئات المجموعات, المثال, التجريد, التفويض, الواجهات, وراثة الواجهات, التصنيف الفرعي. 
المبدأ الأول: الفئات Classes .
كما قلنا سابقاً بأنها الوحدة الأساسية لبناء الكائنات, وقد سبق شرحها بما فيه الكفاية. وستعرف لاحقاً كيفية إنشاء الفئات الخاصة بك.
المبدأ الثاني: الكائنات Objects .
وكما عرفت سابقاً بأن كل ما تراه أمامك في العالم الحقيقي الواقعي يعتبر كائناً, مثل: شخص, أو مكان, أو فاتورة, أو شركة, ...... الخ. ولا تنس أيضاً أن قاعدة البيانات عبارة عن كائن وتحتوي بداخلها على كائنات (جداول - استعلامات - نماذج - عناصر تحكم - تقارير - وحدات نمطية - مجموعات السجلات - ... الخ). وستعرف لاحقاً كيفية إنشاء كائنات خاصة بك.
المبدأ الثالث: التغليف Encapsulation .
وهو باختصار إخفاء طريقة التنفيذ أو الإنجاز عن المستخدم, بحيث لايستطيع المستخدم أن يعدل البيانات الداخلية لكائن ما بطريقة مباشرة, وإنما عن طريق دوال الكائن الداخلية عبر واجهة استخدام خاصة يضعها مبرمج الفئة ويمكن من خلالها فقط تعديل وتغيير بيانات الكائن الداخلية أوخصائصه أوحالته. تعتبر الفئات أحد أشهر الطرق لتغليف العناصر حيث يتم فيها إخفاء تفاصيل تنفيذ أكواد الفئات مما يمنع من أن تستخدم الفئات بطرق غير مرغوب بها, ويسمح لك بالتعديل لاحقاً على مثل هذه العناصر بدون وجود خطر عدم التوافق.
وفي مصطلحات الـ OOP فإن التغليف يسمح للمبرمجين بتخفيض تعقيدات الأكواد عبر إخفاء البيانات والتركيبات المعقدة بداخل الفئات. وسأضرب لكم هذا المثال لتتوضح فكرة التغليف بشكل أكبر:
لنفرض أن لديك صديقاً من هواة برمجة الفيجوال بيسيك للتطبيقات, ويستطيع التعامل جيداً مع الأكواد, لكنه لا يعرف شيئاً عن تقنية الادو, وقد اضطر يوماً ما لأن يتعامل مع هذه التقنية فلجأ إليك لأنك خبير بهذه التقنية وطلب منك مساعدته في ذلك. فقررت أن تساعده بإنشاء فئة تنجز جميع تفاصيل برمجة الادو وتعمل اتصالاً مع قاعدة بيانات لكي تسهل برمجة الادو على صديقك, فقمت بإنشاء فئة وأسميتها CDbConnection والتي تأخذ على عاتقها شتى أوجه الاتصال مع قاعدة البيانات, وتقوم بإنشاء كائنات اتصال, وقد قدمت هذه الفئة طرقاً للاتصال وإغلاق الاتصال مع قاعدة البيانات, وطرقاً لجلب البيانات, وبعد اكتمال التصميم كانت الفئة تشبه الكود التالي:
Option Compare Database
Option Explicit

Private cnn As New ADODB.Connection
Dim p_rst As ADODB.RecordSet

Public Sub GetSql(ByVal strSource As String)
Set p_rst = New ADODB.RecordSet
With p_rst
Set .ActiveConnection = cnn
.Source = strSource
.LockType = adLockOptimistic
.CursorType = adOpenKeyset
.CursorLocation = adUseClient
.Open
End With
End Sub

Public Property Get RecordSet() As ADODB.RecordSet
Set RecordSet = p_rst
End Property

Public Property Set RecordSet(recSource As ADODB.RecordSet)
Set p_rst = recSource
End Property

Public Sub OpenCon(strPath As String)
With cnn
.Provider = "Microsoft.Jet.OLEDB.4.0"
.Open strPath
End With

Exit Sub
End Sub
Public Sub CloseCon()
p_rst.Close
Set p_rst = Nothing
cnn.Close
Set cnn = Nothing
Exit Sub
End Sub
Public Property Get ConObject() As Variant
'Return an object reference of the Connection object.
Set ConObject = cnn
End Property
هذه الفئة تقوم بتغليف برمجة الادو المطلوبة لإدارة الاتصال مع قاعدة بيانات, وبعد ذلك ليس على صديقك سوى وضع هذه الفئة في برنامجه, واستخدامها بالشكل التالي مثلاً:

كود:
Private Sub Form_Load()
    Dim db As New CdbConnection
    db.OpenCon (CurrentProject.Path & "\myDatabase.mdb")
    db.GetSql ("Select * From tbl1")
    Set Me.RecordSet = db.RecordSet
    db.CloseCon
End Sub

	لاحظ ما يلي:
- أن الصديق أو المستخدم للفئة لايعرف طريقة تنفيذ الاتصال, وربما لن يعرف ذلك على الإطلاق. (وهذا ما يسمى بالتغليف والذي هو كما قلنا سابقاً إخفاء طريقة التنفيذ عن المستخدم). 
- أن الفئة قابلة لإعادة الاستخدام مع أي برنامج, وليست مرتبطة ببرنامج معين.
- لا يستطيع الصديق أو المستخدم تعيين قيمة للمتغيرات الخاصة بالفئة بشكل مباشر, لكنه يستطيع ذلك عبر إجراءات الفئة نفسها التي وضعها مبرمج الفئة, مثال: لا تستطيع كمستخدم للفئة أن تسند أي قيمة للمتغير p_rst مباشرة, بل يتوجب عليك أن تتعامل مع إجراءات الفئة أولاً, والسبب يعود لأن المتغير p_rst من نوع Private وبالتالي يصبح محمياً من تعديل المستخدمين, وهذه ميزة قوية في مفهوم التغليف لحماية البيانات أو المتغيرات من عبث المستخدمين.
ملاحظة: لا تقلق إن لم تفهم بعض أو كل ما سبق, فسيتم شرح ذلك فيما بعد, وستبدو هذه الكلمات -وقتئذ- ذات معنى أكثر وضوحاً من الآن. وهذه الأمثلة لمجرد توضيح الموضوع فقط الذي هو هنا "التغليف". في الموضوع القادم -إن شاء الله- سنتطرق إلى المبدأ أو المفهوم الرابع من مبادئ البرمجة الموجهة للكائنات وهو مفهوم الوراثة, وربما نكمل أيضاً المفهوم الخامس إلى جانبه, فإلى الملتقى. 

	المبدأ الرابع: الوراثة Inheritance .
هذا هو الموضوع الرابع في سلسلة البحث المسمى: 
عرفنا في الموضوع السابق أن للبرمجة الموجهة للكائنات خمسة مفاهيم أو مبادئ رئيسة وهي: الفئات, والكائنات, والتغليف, والوراثة, وتعدد الأشكال. وكما سبق أن عرفنا مفهومي الفئات والكائنات في الموضوعين الأول والثاني, عرفنا في الموضوع الثالث مفهوم التغليف وقلنا أنه يعني إخفاء تعقيدات الأكواد عن المستخدم, ومنعه من الوصول إلى المتغيرات الخاصة بالفئة, والسماح له فقط باستخدام ما هو ضروري لإنجاز عمله على أكمل وجه. واليوم نتابع شرح المفهوم الرابع للبرمجة الموجهة للكائنات وهو الوراثة Inheritance .


الوراثة في البرمجة الموجهة للكائنات تعني السماح لإحدى الفئات بأن ترث أعضاء فئة أخرى. (راجع الموضوع الأول من هذا البحث لتعرف ماهي أعضاء الفئة), وهناك نوعان من الوراثة:
1- وراثة الواجهات.
2- وراثة الأكواد التنفيذية.
ولا يخلو مرجع من مراجع الفيجوال بيسيك للتطبيقات من هذه الجملة "للأسف الشديد فإن الفيجوال بيسيك للتطبيقات لا تدعم الوراثة ولا تعدد الأشكال كما تدعمهما اللغات الأخرى.", وهذا صحيح .. ولكن على الرغم من أن الفيجوال بيسيك للتطبيقات لاتدعم من الوراثة سوى وراثة الواجهات, فإن هناك مقاربات لبعض المبرمجين تجعل الفيجوال بيسيك للتطبيقات وكأنها تدعم الوراثة وتعدد الأشكال بشكل كامل (كما سنرى من خلال الأمثلة القادمة في هذا البحث إن شاء الله).

وراثة الواجهات Interface inheritance :

من أجل أن نفهم ماذا تعني وراثة الواجهات, يجب أن نفهم أولاً ماذا يعني مصطلح الواجهات في البرمجة الموجهة للكائنات, وماذا تعني الفئات المجردة أو ما تسمى فئات الواجهة.

ماهي الواجهات Interfaces ؟ :

الواجهات هي تلك الخصائص والطرق والمتغيرات التي يتم الإعلان عنها في الفئة بأنها عامة Public , وبالتالي تصبح هذه الخصائص والإجراءات والمتغيرات مرئية ليس فقط داخل الفئة, ولكن أيضاً خارجها, حيث يستطيع مستخدم الفئة أن يرى أي خاصية أو دالة أو روتين فرعي تم التصريح عنه بأنه Public, بينما لو تم التصريح داخل الفئة عن خاصية أو دالة أو روتين فرعي أو متغير بأنه Private فإنه لن يكون مرئياً سوى داخل الفئة فقط, وبالتلي لا تعتبر هذه الخاصية أو الإجراء بأنها واجهة.

مثال عن الواجهات:

أرجو أن تتذكر أننا أنشأنا في الموضوع السابق فئة أسميناها CdbConnection وقمنا بنمذجتها في حدث عند التحميل لنموذج ما. الآن لو عدت إلى ذلك الحدث وكتبت التالي:


كود:
Private Sub Form_Load()
  
    Dim db As New CdbConnection    .
فستظهر لك بمجرد كتابة النقطة قائمة منسدلة تحوي جميع الإجراءات التي قمنا بكتابتها في الفئة والتي تم الإعلان عنها في الفئة بكلمة Public وهذه الإجراءات والخصائص التي تظهر في القائمة المنسدلة ماهي إلا واجهات الكائن أو واجهات الفئة, بينما سنجد أن المتغير المسمى p_rst والذي تم الإعلان عنه في الفئة بـ Dim لا يظهر في القائمة المنسدلة بالرغم من أنه تم الإعلان عن المتغير في قسم التصريحات العامة للفئة, والسبب هنا يعود لأن Dim تساوي Private .. وما دام المتغير أو الخاصية أو أي دالة أو إجراء داخل الفئة مسبوقاً بالكلمة Private , فلن يكون مرئياً خارج الفئة, وبالتالي لا يعتبر من ضمن واجهات الفئة, إن واجهات الفئة إلا اللآئي تم الإعلان عنها في الفئة مسبوقة بالكلمة Public .. الآن حاول في الفئة السابقة أن تغير كلمة Dim إلى Public ثم عد إلى حدث عند تحميل النموذج وشاهد هل سيظهر المتغير p_rst في القائمة المنسدلة أم لا؟!!

ملاحظة: اتفق جمهور المبرمجين على استخدام الكلمة Private بدلاً من Dim عندما يريدون التصريح عن متغير خاص داخل الفئات (في قسم التصريحات العامة للفئة خاصة), وقد قمت بالتصريح بـ Dim في المثال السابق فقط لكي تعرف أن ذلك ليس حراماً ولا عيباً في التصريح.

ماهي الفئات المجردة Abstract Classes أو فئات الواجهة Interface Classes :

يطلق عليها بعض المبرمجين اسم الفئات المجردة, بينما يطلق عليها البعض الآخر اسم فئات الواجهة (هذا ما أعتقده حيث أعتقد أنهما اسمان مختلفان لمسمى واحد, وأرجو أن تصوبوا لي هذا الاعتقاد إن كان خاطئاً) .. المهم أن هذه الفئات عبارة عن فئات تحتوي -غالباً- على خصائص وربما أيضا أحداث وطرق (دالات وروتينات فرعية), لكنها تخلو من أي كود تنفيذي وإنما مجرد الهيكل الخارجي للإجراءت والخصائص, بحيث تشبه الفئة المجردة أو فئة الواجهة الشكل التالي:

كود:
Option Compare Database
Option Explicit
Public Function ExportFile() As Boolean
End Function
Public Function ImportFile() As Boolean
    
End Function 
Public Property Get FName() As String    
End Property    
Public Property Let FName(ByVal strFileName As String)    
End Property
فإذا أردنا لفئة من الفئات أن ترث واجهات فئة أخرى فإننا نستخدم الكلمة المفتاحية Implements .. وسنتعرف قريباً -إن شاء الله- كيف ومتى يستخدم هذا النوع من الوراثة, مع مثال عملي لتقريب ذلك من الأذهان, كما سنتطرق حينذاك إلى المقاربات التي وضعها مبرمجوا الـ vba لوراثة الأكواد التنفيذية.
في الموضوع القادم إن شاء الله سنتعرف على المفهوم الخامس والأخير من مبادئ البرمجة الموجهة للكائنات وهو مفهوم تعدد الأشكال أو تعدد الواجهات فإلى الملتقى إن شاء الله.
المبدأ الخامس: تعدد الأشكال Polymorphism .
هذا هو الموضوع الخامس في سلسلة البحث المسمى: إنشاء الفئات والبرمجة الموجهة للكائنات
عرفنا في الموضوع السابق أن الوراثة نوعان: وراثة واجهات, ووراثة أكواد تنفيذية, وعرفنا أن الاكسس يدعم وراثة الواجهات فقط, وعرفنا ماهي الواجهات, وقد ابتسرت موضوع الوراثة ابتساراً, وأخرت -عمداً- شرح تفاصيله إلى أجل قريب حتى لا أدخلكم في مجاهيل وطرق تستوحشونها الآن, فالإنسان سلم لما يعرفه, ولما لا يعرفه حرب.

تعدد الأشكال أو تعدد الأوجه Polymorphism :

يسمح تعدد الأشكال لفئة ما أن تنجز نفس العملية أو الإجراء بطرق مختلفة عن فئة أخرى .. كيف ذلك؟.
لنفرض أن لديك برنامجاً لمكتب سفريات داخلية, وعرفت بأن هذا المكتب ينظم سفريات برية, وجوية, وبحرية, فقمت بعد وضع تصميم البرنامج بإنشاء فئة اسميتها الرحلة ITrip حيث قمت فيها بوضع الخصائص والطرق العامة لأي رحلة يمكن أن تحدث, ومن هذه الطرق إجراء يقوم بحساب تكلفة الرحلة. وكان الكود فيها يشبه الكود التالي:

كود:
Option Compare Database
Option Explicit
Function TripCost(iDistance As Integer) As Currency
End Function
وبما أن الرحلة يمكن أن تكون براً عبر السيارة مثلاً, أو جواً عبر الطائرة, أو بحراً عبر سفينة, فقد جعلت فئة الرحلة فئة مجردة من الأكواد أو فئة واجهة, يمكن أن يتم استخدامها في فئة السيارة, أو فئة الطائرة, أو فئة السفينة. ثم قمت بإنشاء فئة السيارة CCar , وجعلت فئة السيارة ترث فئة الرحلة عبر الكلمة المفتاحية Implements بالشكل التالي:

كود:
Option Compare Database
Option Explicit
Implements ITrip
Private Function  ITrip _TripCost (iDistance As Integer) As Currency
    ITrip _TripCost =  iDistance * 150
End Function
ووضعت فئة أخرى أسميتها طائرة CPlane وجعلتها أيضاً ترث فئة الرحلة كما ورثتها فئة السيارة من قبل, وكان كود فئة الطائرة يشبه الشكل التالي:

كود:
Option Compare Database
Option Explicit
Implements ITrip
Private Function  ITrip _TripCost (iDistance As Integer) As Currency
    ITrip _TripCost =  iDistance * 450
End Function
ثم أردت أن تختبر الفئات السابقة فأنشأت وحدة نمطية قياسية ووضعت فيها هذا الكود:
كود:
Sub TestPolyMorphism()
    Dim NewTrip As ITrip
    'The trip by car
    Set NewTrip = New CCar
    MsgBox "Trip by car costs : " & NewTrip.TripCost(50)
    'The trip by plane
    Set NewTrip = New CPlane
    MsgBox " Trip by plane costs: " & NewTrip.TripCost(50)
    Set NewTrip = Nothing
End Sub
مما سبق يتضح أن الوظيفة TripCost قد أعطتنا نتيجتين مختلفتين لنفس المسافة المقطوعة, بسبب أن كل فئة قد استخدمت نفس الوظيفة بطريقة مختلفة عن الفئة الأخرى, وهذا هو باختصار ما يسمى بتعدد الأشكال, أو تعدد الأوجه.
ستتعرف - عند تقدمنا في هذا البحث - على المزيد من المعلومات والتفاصيل الدقيقة حول الوراثة في المواضيع القادمة إن شاء الله. وبهذا نكون قد أخذنا فكرة عن مبادئ البرمجة الموجهة للكائنات, وفي الموضوع القادم - إن شاء الله - سنتعرف على كيفية إنشاء الفئات, فإلى الملتقى.
إنشاء خصائص الفئات

هذا هو الموضوع السابع في سلسلة البحث المسمى: إنشاء الفئات والبرمجة الموجهة للكائنات
عرفنا في الموضوع السابق أسس بناء الفئات -ولنا إليها عودة عن قريب إن طال بنا العمر, كي نشرحها بالتفصيل-, وعرفنا كيف ننشئ الفئات, وكيف نصرح عن متغيرات الفئة الداخلية, واليوم سنعرف كيفية إعطاء المستخدم طريقةً لقراءة وكتابة قيم هذه المتغيرات.

بين خصائص الفئة Class Properties وسمات الكائنات Object attributes :

لا تكون معظم الفئات ذات قيمة مالم يكن لها خصائص يمكن أن نقوم بإعداد وتعديل قيمها أو أن نسترجع قيم هذه الخصائص عندما تدعو الحاجة إليها. تستطيع الخصائص Properties أن تخزن بداخلها قيماً لمتغيرات الفئة الداخلية التي صرحنا عنها سابقاً, حيث تمثل قيم هذه المتغيرات ما يسمى بـ سمات الكائن Object attributes . وبذلك يمكن لكل خاصية من الخصائص أن يكون لها عدة سمات تبعاً لقيمة ذلك المتغير, فسمة خاصية الطول -مثلاً- لدى الكائن عارف حسان هي معتدل, بينما سمة خاصية الطول لدى الكائن محمد عارف هي طويل. لكن الكائن الواحد لا يمكن أن تكون له أكثر من سمة واحدة لنفس الخاصية, وهذا أمرٌ بدهي, فلا يمكن -مثلاً- لكائن ما أن تكون سمة الطول لديه طويل قصير في آن معاً .. أليس كذلك؟.

إنشاء خصائص الفئة:

بعد أن قمنا بتعريف متغيرات الفئة الداخلية, فإننا نحتاج لإنشاء آلية لقراءة أو كتابة هذه القيم, من أجل هذا يجب أن نقوم بتعريف الخصائص, وهناك طريقتان لإنشاء الخصائص, الطريقة الأولى وهي الأسهل وذلك عن طريق الإعلان في قسم التصريحات العامة للوحدة النمطية للفئة عن متغير عام Public . مثل:
كود:
· Public Value As Long
حيث يصبح هذا المتغير - بمجرد الإعلان عنه - خاصية من خصائص الفئة, كما يصبح واجهة للفئة يمكن لأي إجراء خارج الفئة أن يراها وأن يعرف قيمتها, أو أن يعدل هذه القيمة كما يحب ويهوى, وهذا يعني أن هذه الخاصية تكون قابلة للقراءة والكتابة, واسم المتغير -في هذه الحالة- يصبح هو اسم الخاصية, ويكون مرئياً للإجراءات المستهلكة للفئة حيثما ثقفت(مستهلكات الفئة Class Consumers : هي الإجراءات التي تكون خارج الفئة, ويتم فيها استعمال الفئة), لذلك إن كنت ستستخدم هذه الطريقة لإنشاء الخصائص -وهي غير محبذة- فاختر اسماً ذا معنىً للمتغير يكون معبراً عن الخاصية التي يمثلها.
وعلى الرغم من سهولة هذه الطريقة في تعريف خصائص الفئات, إلا أن لها عدة مساوئ:
1- إذا قام مستخدم الفئة بتغيير قيمة الخاصية, فلا تستطيع الفئة أن تعرف ما هي البيانات التي قام المستخدم بإدخالها, وبالتالي لا تستطيع ضبط هذا التغيير, فربما يقوم المستخدم بإدخال نوع بيانات مختلف, وقد يكون هذا الضبط ضرورياً أحياناً في حالات مثل: تقييد القيم المدخلة ضمن مجال معين من القيم, أو اتخاذ إجراء معين عند تغيير قيمة الخاصية.
2- لا تستطيع وضع قواعد التحقق من صحة البيانات, أو تقييد قيم الخاصية, على سبيل المثال: ربما قد ترغب أن تقيد خاصية تمثل عمر الشخص بأرقام حقيقية موجبة فقط.
3- لا تستطيع عبر هذه الطريقة أن تنشئ خصائص للقراءة فقط Read Only Property , فالخاصية التي يتم إنشاؤها بهذه الطريقة تكون للقراءة والكتابة Read\Write , وبالتالي يمكن لمستخدم الفئة أن يلهو بها كما يشاء, وأحياناً يكون من المهم في برنامجك أن يحصل المستخدم على قيمة الخاصية دون أن تكون له قدرة على تغيير هذه القيمة.
ماهي إجراءات الخاصية Property Procedures ولماذا نستخدمها؟:

من أجل أن نتجنب المساوئ السابقة, فإننا نلجأ لما يسمى بإجراءات الخاصية Property procedures , وهي عبارة عن إجراءات vba يتم تنفيذها عندما يتم إسناد أو استرجاع قيمة لخاصية من الخصائص. بإمكانك أثناء برمجة إجراء خاصية أن تقوم بكتابة أوامر برمجية تنفذ أفعالاً معينة في ذلك الإجراء.
لإجراءات الخاصية ثلاثة أنواع وهي Property Get و Property Let و Property Set . الإجراء الأول Property Get مهمته أن يعيد لنا قيمة الخاصية, سالكاً بذلك سلوك الدوال المعروف بإعادتها للقيم, بينما المهمة المنوطة بالإجرائين الآخرين Property Let و Property Set هي إعداد قيمة الخاصية, والفرق بين إجراء الخاصية Let وإجراء الخاصية Set هو أن إجراء الخاصية Let يستخدم مع القيم من جميع الأنواع(نص - عدد صحيح - عدد مفرد - تاريخ/وقت - .... الخ) ماعدا القيم التي تكون من نوع كائن Object ففي هذه الحالة نستخدم معها إجراء الخاصية Set . وبالمثال يتضح المقال.
في الموضوع السابق قمنا بإنشاء فئة أسميناها CSample وقمنا فيها بالتصريح عن بعض المتغيرات الخاصة, والآن سنقوم بإنشاء إجراءات خصائص لتلك الفئة, فلنبدأ بسم الله:
1- قم بفتح الفئة السابقة, ومن شريط أدوات محرر vba انقر فوق Insert إدراج.
2- من القائمة المنسدلة انقر فوق Procedure لفتح مربع حوار Add Procedure إضافة إجراء.
3- قم بكتابة اسم الخاصية في مربع نص Name . وليكن اسمها في هذا المثال Name .
4- من مجموعة الاختيار Type اختر Property , ثم انقر زر OK .
ستجد أن الاكسس قد قام بشكل تلقائي بإضافة إجرائين للخاصية Name , أحدهما إجراء خاصية من نوع Get ومهمته قراءة قيمة الخاصية, الآخر إجراء خاصية من نوع Let ومهمته إسناد قيمة الخاصية, وتكون نوع بيانات الإجرائين بشكل افتراضي من نوع Variant , وهنا يجب تغيير نوع بيانات الخاصية لكي تطابق نوع بيانات المتغير الداخلي الخاص للفئة وهو هنا mSName وهو من نوع نص String .
5- يمكنك كتابة أي كود تراه مناسباً لإي إجراء من إجراءات الخاصية مثل كتابتك لأي إجراء أو دالة في وحدة نمطية قياسية, لكنه ينصح بالبساطة عند كتابة أكواد الخصائص بشكل خاص والفئات بشكل عام, كما يمكنك -إذا استدعى الأمر- أن تضيف أي وسيطة أو معاملات إلى إجراءات الخصائص بكافة أنواعها الثلاثة, وسأشرح ذلك لاحقاً مع الأمثلة إن وجدت إلى ذلك سبيلاً.
6- كرر الخطوات السابقة لإضافة إجراء آخر باسم DOB وقم بتغيير نوع بيانات الخاصية إلى Date لتطابق نوع بيانات المتغير الخاص الداخلي mDOB . مع ملاحظة أنه عند إضافة خاصية جديدة, يجب الانتباه جيداً لموقع مؤشر الإدخال, حتى لا يتم إدراج إجراء الخاصية الجديد داخل إجراء سابق آخر, أو في مكان غير مناسب فلا تعمل الخاصية عند ذلك بشكل صحيح.
إن شئت أن تجعل الخاصية للقراءة فقط أو للكتابة فقط, قم بحذف أحد الإجرائين السابقين, فإذا أردت أن تجعل الخاصية للقراءة فقط, فاحتفظ بإجراء الخاصية Get واحذف إجراء الخاصية Let والعكس صحيح أيضاً. في المثال التالي أضفت الخاصية Age وهي للقراءة فقط, بحيث أصبح الكود الخاص بالفئة CSample يشبه ما يلي:
كود:
Option Compare Database
Option Explicit

Private mValue As Currency
Private mSName As String
Private mDOB As Date

Public Property Get Name() As String
    Name = mSName
End Property

Public Property Let Name(ByVal vNewValue As String)
    mSName = vNewValue
End Property

Public Property Get DOB() As Date
    DOB = mDOB
End Property

Public Property Let DOB(ByVal vNewValue As Date)
    mDOB = vNewValue
End Property

Public Property Get Age() As Byte
    Age = DateDiff("yyyy", mDOB, Date)
End Property
سأدع لك المتغير الخاص mValue لكي تقوم بتعريف خاصية من نوع عملة Currency وليكن اسم الخاصية Salary لنضع فيها الراتب الشهري.
في الموضوع القادم إن شاء الله سنعرف كيف نختبر الفئة الجديدة CSample عبر إنشاء كائنات خاصة بنا تدعى مثيلات الفئة Instances والتي أعتبرها بمثابة القلب النابض للبرمجة الموجهة للكائنات, فهي التي تجعل الفئات مفعمة بالنشاط, و ملأى بالحيوية, حتى ليخيل إليك أن روحاً من أمر الله قد سرت فيها فأصبحت تضج بالحياة وكأنها عرائس من الجن قد أصبحن طوع يمينك, ولولا تلك المثيلات لكانت الفئات كأحجار ملقاة في الفيافي والقفار لا سعر لها ولا قيمة, كما سنتطرق إلى معلومات مهمة أخرى, فإلى الملتقى.
تمثيل الفئات Classes Instantiating
هذا هو الموضوع الثامن في سلسلة البحث المسمى:
إنشاء الفئات والبرمجة الموجهة للكائنات
عرفنا سابقا كيفية إنشاء خصائص الفئات, وأن الطريقة المثلى لذلك تكون عبر إجراءات الخصائص لما فيها -أولاً- من مفهوم التغليف, وثانياً لأنها تجنبنا مساوئ الطريقة الأخرى في إنشاء الخصائص, كما أنها تمنحنا العديد من الميزات. ولعلكم لاحظتم أن إجراءات الخاصية هي الإجراءات الوحيدة في الفيجوال بيسيك التي تستطيع أن تحمل نفس الاسم -لإجرائين مختلفين- ضمن نفس الوحدة النمطية.

إنشاء كائن مثيل للفئة Instance :
يتعرف الفيجوال بيسيك للتطبيقات على الفئة حالما تقوم بحفظها, وعند كتابتك للكلمة As بعد اسم متغير تصرح به, فستجد أن قائمة منسدلة قد ظهرت وبها أسماء جميع أنواع البيانات المتوفرة ومن ضمنها اسم الفئة التي قمت بإنشائها.
لاختبار الفئة الجديدة CSample , قم بإنشاء وحدة نمطية قياسية -(وليس وحدة نمطية للفئة)- لتعرف أنه أصبح بإمكانك الآن إنشاء كائنات خاصة بك تدعى مثيلات الفئة Instances حيث تعتبر هذه المثيلات لب وجوهر البرمجة الموجهة للكائنات وغايتها القصوى.
1- ضع إجراءاً في وحدة نمطية قياسية بأي اسم تريد وليكن testClass . 
2- قم بالإعلان عن متغير -بأي اسم تريد- من نوع كائن الفئة وليكن اسم المتغير myObj . 
سنستخدم هذا المتغير في تخزين مرجع يشير إلى مثيل جديد للفئة. تخضع المراجع التي تشير إلى مثيلات الفئة لنفس القواعد التي تخضع لها المراجع التي تشير إلى الكائنات المبنية في الآكسس. يمكنك التصريح عن المتغير بأي من الكلمات المحجوزة التالية Dim أو Private أو Public أو Global . ثم نأتي لآخر خطوة وهي إنشاء مثيل جديد للفئة, وأن نخزن داخل المتغير مرجعاً يشير إلى المثيل. (للمزيد من المعلومات حول هذه النقطة, راجع كتاب تركي العسيري فيجوال بيسيك للجميع صفحة رقم168 "صورة الكائن في الذاكرة").
3- قم بإنشاء كائن مثيل بواسطة التعليمة Set والكلمة المفتاحية New كما هو موضح في الكود التالي:
Sub testClass()
    Dim myObj As CSample
    Set myObj = New CSample
    myObj.Name = "Aref Hassan"
    myObj.DOB = #9/11/1965#
    MsgBox "My Name is " & myObj.Name & vbNewLine & _
           "I am " & myObj.Age & " years old."
    Set myObj = Nothing
End Sub
لاحظ ما يلي:
1- في السطر الأول من الكود السابق أعلنا عن متغير myObj من نوع وجنس الفئة CSample بحيث أصبح هذا المتغير -الآن- يشير إلى منطقة في الذاكرة تحتوي على تراكيب الفئة(عبارة عن جدول في الذاكرة يخزن فيه تراكيب تعرّف عمل إجراءات وخصائص وأحداث الفئة), ولنسمّ هذه المنطقة من الذاكرة -مثلاً- المنطقة1. 
2- في السطر الثاني استخدمنا الكلمة المفتاحية Set .. والكلمة المفتاحية New لتهيئة المتغير myObj بحيث يصبح كائناً جديداً من نوع وجنس الفئة CSample وهذه العملية تسمى بالتمثيل Instantiating أو الاستنساخ, ويسمى الكائن الناتج عنها بالمثيل Instance .. وبذلك يصبح المتغير myObj كائناً مثيلاً يشبه الفئة ويحمل جميع واجهاتها, ويشير إلى منطقة جديدة New في الذاكرة يمكن أن تخزن القيم التي يدخلها المستخدم, أو التي يمكن أن تنتج عند استعماله لطرق الفئة, ولنسمّ هذه المنطقة من الذاكرة بالمنطقة2. تكون المنطقة2 في الذاكرة على علاقة بالمنطقة1, بحيث أن المنطقة1 يتم حجزها لمرة واحدة فقط مهما تكرر الإعلان عن متغيرات من نفس الفئة, بينما المنطقة2 يمكن أن تتعدد مناطقها في الذاكرة كلما تم الإعلان عن متغير جديد من نفس الفئة, لذلك فمن الحكمة أن تستخدم التعليمة Set أيضاً في نهاية الإجراء لإنهاء الكائنات وضمان محوها من الذاكرة بتحويل قيمتها إلى Nothing .

الكلمتان المفتاحيتان Set و New :

قد يكون استخدام التعليمة Set ضرورياً عند إسناد الكائنات, ولو لم يتم استخدامها فقد تظهر رسائل خطأ أو نتائج غير متوقعة, أما استخدام الكلمة المفتاحية New فهي ضرورية جداً لإنشاء كائن مثيل جديد, ولو لم يتم استخدامها وحاولت استخدام إحدى خاصيات أو طرق الفئة, فستظهر رسالة خطأ أثناء التشغيل تخبرك أن "هناك متغير كائن أو متغير محصور لم تتم تهيئته", وقصارى القول أن السطر الأول لوحده لا يكفي لإنشاء مثيل جديد.

اختصار سطر في الكود قد يكون مكلفاً:

يمكن إنشاء مثيل جديد للفئة من خلال سطر واحد فقط أثناء الإعلان عن المتغير كما يلي:

كود:
Dim myObj As New CSample
ويصبح بإمكانك استخدام خصائص وطرق الفئة مباشرة بعد هذا التصريح, وبدون الحاجة إلى استخدام التعليمة Set وسيتم تمثيل الكائن بشكل تلقائي أثناء تنفيذ الكود, عند أول استخدام لخصائص أو لطرق المثيل, وهذا ما يعتبر ربطاً متأخراً (لأنه لا يتم حجز المنطقة2 في الذاكرة قبل البدء الفعلي في استخدام الكائن), أما الطريقة السابقة التي استخدمنا فيها التعليمة Set فتعتبر ربطاً مبكراً (لأنه يتم حجز المنطقة2 في الذاكرة أثناء ترجمة الكود Compiling Time وليس أثناء تنفيذه, والترجمة تسبق التنفيذ, لذلك يكون الربط المبكر أسرع تنفيذاً بكثير من الربط المتأخر), وتزداد أهمية هذه المعلومة في البرامج الكبيرة, فقد نحتاج -أحياناً- في البرامج المعقدة التي تحتوي على فئات كثيرة أو تحوي فئاتها أكواداً معقدة, إلى أن نعرف متى وأين يقوم الـ vba بإنشاء مثيل جديد للفئة, حيث قد يكون اختصار هذا السطر سبباً في عدم حصولنا على النتائج المرجوة من الأكواد التي تعبنا في كتابتها للبرنامج.
من كان له سؤال حول هذا الموضوع أو المواضيع السابقة, فأرجو أن لا يتردد في طرحه, وسأحاول الإجابة عليه ما استطعت.
في الموضوع القادم - إن شاء الله - سنتطرق إلى كيفية إنشاء الطرق الأخرى في الفئات, بالإضافة إلى معلومات أخرى جديدة فإلى الملتقى.
إنشاء الطرق Methods في الفئات

هذا هو الموضوع التاسع في سلسلة البحث المسمى: إنشاء الفئات والبرمجة الموجهة للكائنات
عرفنا في المواضيع السابقة بأنه حالما يكون لديك متغير يشير إلى مثيل جديد للفئة, فيمكنك عندها استخدام الخصائص والطرق الموجودة في الوحدة النمطية للفئة, كما عرفنا أن طرق الفئة هي جميع الإجراءات Procedures الموجودة في الفئة سواءاً كانت هذه الإجراءات روتينات فرعية Subs أو دالات (وظائف) Functions -(البعض يميل أيضاً إلى اعتبار الخصائص من الطرق, لأنها عبارة عن إجراءات أيضاً)-, وقد سبق شرح إنشاء الخصائص, وسيتم في المواضيع القادمة -إن شاء الله- تقديم المزيد من الأمثلة لتوضيحها وشرحها بشكل أكبر. واليوم سنتعرف على كيفية إنشاء بعض الطرق في الفئة CSample التي مرت معنا سابقاً, كما سنعرف متى نستخدم الطرق, ومتى نستخدم الخصائص.

إنشاء الطرق Creating methods :

عرفنا سابقاً أنه بمجرد التصريح في الفئة عن متغير عام Public فأننا بهذا نكون قد أنشأنا خاصية من خصائص الفئة, كذلك عند إنشاء بقية طرق الفئة, فبمجرد التصريح عن إجراء عام Public procedure فأننا بذلك نكون قد أنشأنا طريقة في الفئة. يمكننا أن ننشئ دالات عامة Public functions و روتينات فرعية عامة Public subs والفرق الوحيد بين الدالات والروتينات الفرعية أن الدالات يمكن أن تعيد قيمة لمن يقوم باستدعائها, بينما الروتينات الفرعية لا تعيد قيمة معينة بل تنفذ أمراً واحداً أو مجموعة من الأوامر البرمجية.

متى نستخدم إجراءات الخاصية, ومتى نستخدم إجراءات الدالة؟:

عرفنا سابقاً بأن إجراء الخاصية Property Get يسلك سلوك الدوال Functions في أنه يعيد إلينا قيمة Value عندما نقوم باستدعائه, وهنا قد يتبادر إلى الذهن سؤال وهو, متى أستخدم الدالة, ومتى أستخدم الخاصية؟.
الجواب: كما قلنا سابقاً أنه يجب عليك عند تصميم الفئة أن تتصور و تتساءل: ماهي أعضاء الفئة الضرورية لإنجاز ما هو مطلوب من الفئة إنجازه؟!! .. وبعد أن يكون لديك تصور عن أعضاء الفئة, يعود الأمر -برمته- إليك -كمبرمج- في أن تقرر أياً من هذه الأعضاء من يجب أن تكون خصائص للفئة, و أياً منها من يجب أن تكون طرقاً لها.
تستخدم الخصائص -في العادة- لتمثل القيم البسيطة التي يتم إنجازها عن طريق متغيرات فردية أو عن طريق عمليات حسابية تعتمد على البيانات الموجودة في الذاكرة, بينما تدل الطرق -من جهة أخرى- على عمل يتطلب كمية لا بأس بها من الأكواد, كما تشير إلى أن الوصول إلى النتيجة المرغوبة يتطلب -ربما- جهداً وعملاً يستحق الإشادة والتقدير.
قد يكون ضرورياً أحياناً أن تنجز كطريقة ما يعتبر عادة خاصية, إذ ربما نلجأ إلى ذلك عندما يكون الحصول على القيمة أو إعدادها يتطلب عملاً أكثر من مجرد الإشارة إلى متغير, على سبيل المثال فإن كتابتك لخاصية تحدد للمستخدم فيما إذا كانت الطابعة تعمل بشكل سليم أم لا قد يستغرق وقتاً ويتطلب العديد من الخطوات, لذلك ربما تختار أن تنفذ هذا الأمر كطريقة عوضاً عن الخاصية, وينطبق هذا الأمر -أيضاً- عندما يتطلب تغيير إعدادات الخاصية أكثر من قيمة واحدة, وفي هذه الحالة تكون الطريقة مع قليل من الوسائط أكثر ملائمة من الخاصية.
نصيحة: عندما تقرر أن تكتب طريقة في نفس الوقت التي تكون فيها هذه الطريقة أقرب إلى مفهوم الخاصية, فاجعل اسم الطريقة يبدأ بـ Get أو بـ Set مثل: GetSetting أو SetThreeDFormat فهذا يعطي معنى لواجهات الفئة ويجعلها أقرب إلى القلب والعقل.
تطبيق عملي:
سنعود إلى الفئة التي قمنا بإنشائها سابقاً, وطلبت منكم أن تضيفوا إليها خاصية باسم Salary , وسنقوم الآن بإضافة طريقة تقوم بحساب الضريبة على الراتب وليكن اسمها مثلاً Tax , بحيث يصبح كود الفئة كما يلي:

كود:
Option Compare Database
Option Explicit
'You can creat a property by declaring a Public variable as follows:
Public Value As Long
'====================
Private mValue As Currency
Private mSName As String
Private mDOB As Date

Public Property Get Name() As String
    Name = mSName
End Property

Public Property Let Name(ByVal vNewValue As String)
    mSName = vNewValue
End Property

Public Property Get DOB() As Date
    DOB = mDOB
End Property

Public Property Let DOB(ByVal vNewValue As Date)
    mDOB = vNewValue
End Property

Public Property Get Age() As Byte
    Age = DateDiff("yyyy", mDOB, Date)
End Property

Public Property Get Salary() As Currency
    Salary = mValue
End Property

Public Property Let Salary(ByVal vNewValue As Currency)
    mValue = vNewValue
End Property
Public Function Tax() As Currency
    If Me.Salary <= 100 Then
        Tax = 0.1 * Me.Salary
    Else
        Tax = 0.12 * Me.Salary
    End If
End Function
ولكي نختبرالفئة بعد هذا التعديل, نضيف الكود التالي إلى وحدة نمطية قياسية ثم نقوم بتشغيله:


Sub testClassMethod()
    Dim myObj As CSample
    Set myObj = New CSample
    myObj.Name = "Aref Hassan"
    myObj.DOB = #9/11/1965#
    myObj.Salary = 350
    MsgBox "My Name is " & myObj.Name & vbNewLine & _
           "I am " & myObj.Age & " years old." & vbNewLine & _
           "My salary is " & myObj.Salary & vbNewLine & _
           "The tax of salary = " & myObj.Tax
    Set myObj = Nothing
End Sub
اضطررت إلى تبسيط أكواد الخصائص والطرق حتى يستطيع المبتدئ والمتوسط أن يفهم الفكرة من الموضوع, وبعد أن نكمل شرح المواضيع الأساسية -بإذن الله- ستصبح أكواد الأمثلة - بشكل أكثر - موجهة للمحترفين, لذلك أرجو من أساتذتي المحترفين أن يصبروا قليلاً.
في الموضوع القادم -إن شاء الله- سنتطرق إلى أحداث الفئات بالإضافة إلى عناوين أخر فإلى الملتقى.
أحداث الفئات المبنية
هذا هو الموضوع العاشر في سلسلة البحث المسمى: إنشاء الفئات والبرمجة الموجهة للكائنات
عرفنا في المواضيع السابقة كيف نقوم بإنشاء الفئات بخصائصها و طرقها, وكيف نقوم بإنشاء مثيلات الفئة, وعرفنا كذلك أن هذه المثيلات تشكل الفرق بين الوحدات النمطية للفئات والوحدات النمطية القياسية, وهذا صحيح, لكن هناك فرقاً آخر بينهما تعمدت عدم ذكره سابقاً حتى لا تشعروا - في مرحلة مبكرة - بالوحشة منه, وهو أن للوحدات النمطية للفئات حدثان يميزانها عن الوحدات النمطية القياسية وهما: حدث عند البدء Initialize, وحدث عند الختام Terminate واللذان يرافقان -على الترتيب- لحظة إنشاء مثيل جديد للفئة و لحظة إنهاءه وتدميره وموته.

الحدثان Initialize و Terminate :

تحتوي كل الفئات بشكل افتراضي على حدثين مبنيين هما Initialize و Terminate ويمكن الوصول إليهما كما يلي:
عندما تكون في وحدة نمطية للفئة وحركت بصرك إلى أعلى صفحة المحرر, ستجد أن هناك مربعي تحرير وسرد, يقع الأول إلى الأعلى والأيسر, ويقع الثاني إلى الأعلى والأيمن .. اختر من القائمة المنسدلة العليا اليسرى الكائن Class وستجد أن الفيجوال بيسيك للتطبيقات قد أدرج لك إجراء الحدث Initialize بشكل تلقائي, وستجد في القائمة المنسدلة العليا اليمنى وجود الإجرائين Initialize و Terminate حيث يمكنك النقر على أحدهما لإضافته عند الرغبة.
يشبه حدث Initialize ما يسمى بالمشيدات Constractors في لغات الـ OOP الأخرى, ويتم قدحه عندما يتم إنشاء مثيل جديد للفئة لأول مرة, بينما يتم قدح الحدث Terminate عند إنهاء وتدمير آخر مرجع يشير إلى ذلك المثيل. كيف؟ .. هل يمكن أن يشير أكثر من مرجع إلى نفس المثيل؟ .. الجواب: نعم, يمكن أن يكون للمثيل الواحد أكثر من مرجع يشير إليه كما يلي:

كود:
Dim X As CSample
Dim Y As CSample
Set X = New CSample
Set Y = X
يبرز في الكود السابق سؤالان وهما: هل لدينا مثيل واحد للفئة أم مثيلان مختلفان؟ .. هل سيتم تدمير الكائن Y إذا قمنا بتدمير الكائن X ؟.
الجواب: لدينا كائنان أو مرجعان مختلفان يشيران إلى مثيل واحد فقط, لكن تدمير وإنهاء وإزالة أحدهما من الذاكرة لا يعني تدمير وإنهاء وإزالة الآخر, بله حتى لا يعني إزالة المثيل نفسه من الذاكرة, حيث يبقى المثيل متواجداً في الذاكرة طالما هناك مرجع واحد -على الأقل- يشير إليه, فإذا زالت جميع المراجع التي تشير إلى ذلك المثيل, عندها يزول المثيل ويمحى من الذاكرة ويصبح مدمراً بلا بواقي, ميتاً بلا بواكي, كأن أمه الفئة لم تلده, أوكأن أباه المبرمج لم يُردْهُ.

فيم يستخدم الحدثان؟:

لعلكم تذكرون أننا قلنا في موضوع "إنشاء مثيلات الفئة" أنه من المهم معرفة متى يتم إنشاء مثيل جديد للفئة, ومتى يتم إنهاء هذا المثيل, لأنه تكون لدينا الفرصة لتشغيل أكواد برمجية استجابة لأي حدث منهما.
يمكنك استغلال حدث Initialize في إعداد القيم الافتراضية للخصائص, أو في إنشاء مراجع للكائنات المحلية الأخرى (المقصود بالكائنات المحلية: هي المتغيرات من نوع كائن أو فئة والتي يتم الإعلان عنها واستخدامها داخل الفئة). أما الحدث Terminate فغالباً ما يستخدم لإنجاز مهمات التنظيف, وتحرير الذاكرة من مراجع تلك الكائنات المحلية.
كذلك يمكن استخدام الحدثين Initialize و Terminate في عملية تتبع الأكواد التي تقوم بكتابتها debugging فإذا أردت معرفة لحظة ولادة أحد كائناتك الخاصة, أو لحظة موته وفناءه, فالجأ إلى ذينك الحدثين الجللين كما يلي على سبيل المثال:

Private Sub Class_Initialize()
    Debug.Print "Object created."
End Sub

Private Sub Class_Terminate()
    Debug.Print "Object destroyed."
End Sub
تنصح شركة ميكروسوفت بعدم استخدام دالة الرسائل MsgBox في حدثي Initialize و Terminate بشكل خاص, وقد قرأت سابقاً -لا أتذكر متى ولا أين, وهذه ضريبة التقدم في العمر- بأنه ينصح بعدم استخدام دالة الرسائل في أكواد الفئات بشكل عام, ويفضل أن يستعاض عنها بإنشاء أحداث خاصة بك يتم قدحها عند حدوث ما يوجب ظهور تلك الرسائل, وهذا ما سنتناوله في المواضيع القادمة إن شاء الله.

تطبيق عملي:

إلى جانب الفئة CSample التي أنشأناها سابقاً, قم بإنشاء فئة جديدة واحفظها باسم CBEvents وضع فيها الكود التالي:

كود:
Option Compare Database
Option Explicit

Private myObj As CSample
Private mlngDefault As Long

Private Sub Class_Initialize()
    Set myObj = New CSample
    mlngDefault = 1234
End Sub


Private Sub Class_Terminate()
    Set myObj = Nothing
End Sub

Public Property Get Name() As String
    Name = myObj.Name
End Property

Public Property Let Name(ByVal vNewValue As String)
    myObj.Name = vNewValue
End Property

Public Property Get setDefaultValue() As Long
    setDefaultValue = mlngDefault
End Property

Public Property Let setDefaultValue(ByVal vNewValue As Long)
    mlngDefault = vNewValue
End Property
ولاختبار الفئة الجديدة, ضع الإجراء التالي في وحدة نمطية قياسية, ثم قم بتشغيله:

Sub testClassBEvents()
    Dim objE As CBEvents
    Set objE = New CBEvents
    objE.Name = "Aref Hassan"
    MsgBox objE.Name & vbNewLine & _
            "Dfault Value = " & objE.setDefaultValue
    Set objE = Nothing
End Sub
في الموضوع القادم -إن شاء الله- سنتعرف على الفئات الهرمية, وفئة المجموعة, فإلى الملتقى.
التسلسل الهرمي للفئات Class Hierarchies
هذا هو الموضوع الحادي عشر في سلسلة البحث المسمى: إنشاء الفئات والبرمجة الموجهة للكائنات
عرفنا في الموضوع السابق أن الفئات تحتوي بشكل افتراضي على حدثين وهما Initialize و Terminate , وعرفنا ما هي الاستخدامات الممكنة لكل منهما, واليوم سنتعرف على التسلسل الهرمي للفئات, وعلى علاقاتها فيما بينها, وكيفية استخدام إحدى الفئات داخل فئة أخرى.

ما هي الفئات الهرمية:
عندما تصبح ملماً بأساسيات استخدام الوحدات النمطية للفئات, وتصبح الفئات لديك أكثر عدداً وتعقيداً, فلابد أن تعرف حينئذ كيف تستخدم هذا العدد من الفئات بشكل ناجح وفعال في بناء تطبيقات معقدة تعتمد في بنائها على الفئات. وقد عرفت -في المواضيع السابقة- أن الفئات عبارة عن كائنات, وأنه يمكن للكائنات أن تحوي بداخلها كائنات أخرى, وبالتالي يمكن للفئة أن تحوي داخلها فئات أخرى, وأعتقد أن هذا هو ما يعطي المرونة والقابلية للابتكار في البرمجة الكائنية, ويدخل مفهوم الفئات الهرمية في هذا السياق, وكما قلنا سابقاً بأنه يجب على مبرمج الفئات عند تصميم التطبيق أن يتصور الفئات التي سيحتاجها البرنامج قبل أن يتصور الأكواد الضرورية له, وتلك المجموعة من الفئات التي تعتقد أنها ضرورية لبناء برنامجك هي ما تسمى بالفئات الهرمية.
الفئات الهرمية عبارة عن عدة فئات (عنقود من الفئات) تكون مرتبطة مع بعضها بعضاً بحيث يبدو هذا الارتباط وكأن هناك فئة تقف على رأس هرم من الفئات الأخرى وتسمى هذه الفئة بالفئة الأم أو الفئة الأساس Base Class بينما تسمى الفئات الأخرى المتعلقة بها بالفئات البنات أو الفئات المشتقة Derived Classes والتي تشكل قاعدة الهرم. قد يحتوي التطبيق الواحد على العديد من الفئات الهرمية, هذا التجمع من الفئات والتي يتم سلكها في نظام واحد هو ما يشكل ما يسمى بكائن المثال object model والذي يقودنا -بالضرورة- إلى الحديث عن العلاقات بين الفئات, وإلى نوع من وراثة الفئات يسمى الوراثة بالتفويض, وإلى فئات المجموعات والتي قد تحتاج إلى أكثر من موضوع للحديث عنها.

كيفية تمثيل العلاقات بين الفئات:
العلاقات بين الفئات -بشكل عام- ليست كالعلاقات بين الجداول, إذ أن العلاقات بين الجداول أشد صرامة, وأكثر قوة ومتانة, ولكي تنشئ علاقة بين الفئة الأم و فئة مشتقة -أو بين أي فئتين-, ما عليك سوى أن تصرح عن متغير من نوع الفئة الأم في بداية الفئة المشتقة, وبهذه السهولة تكون قد أنشأت علاقة واحد إلى واحد بين الفئتين .. أما إذا أردت أن تنشئ علاقة واحد إلى متعدد, فالأمر متاح أيضاً عن طريق وضعها في مجموعة collection أو إنشاء مصفوفة من الفئات, أو عن طريق إنشاء فئة مجموعة, وهذا ما سنتحدث عنه في الموضوع القادم إن شاء الله.

الوراثة بالتفويض Delegation :

تحتوي الفئة المشتقة -كأي فئة أخرى- على خصائص وطرق خاصة بها بالإضافة إلى أنها يمكن أن تحتوي على كل أو بعض أعضاء -(خصائص و طرق)- الفئة الأم (وهذا ما يعتبره البعض نوعاً من الوراثة يسمى الوراثة بالتفويض).
الفئة المشتقة -نفسها- يمكن أن تكون أمّاً لفئات أخرى مشتقة منها, وبالتالي يصبح هناك هرمٌ جديدٌ داخل الهرم السابق, وتصبح الفئة الأمُّ الأولى جدةً لهذه الفئات الجديدة واللاتي يمكن أن يرثن أعضاء من أمهن إلى جانب أعضاء من جدتهن وأن يحصلن -أيضاً- على أعضاء خاصة بهن لم تكن موجودة في أسلافهن. .. أليس هذا ممتعاً؟.

تطبيق عملي:

لنفرض أنك قمت بتحليل نظام الموظفين لشركة فيها مجموعات متعددة ومختلفة من الموظفين, ولاحظت أنه بالرغم من وجود سمات عامة لجميع المجموعات, إلا أن كل مجموعة من الموظفين تختلف عن الأخرى بعدد ضئيل من الطرق, وقد تم تكليفك ببناء تطبيق بالاكسس يساعد في عمل الحسابات المناسبة لكل مجموعة, فقمت بعمل التالي:
1- إنشاء الفئة الأساس Base Class حيث أن المهمة الأولى هي تحديد السمات العامة المشتركة لدى جميع الموظفين مثل الاسم, والعمر, ووو .... الخ, وكذلك العمليات الحسابية المشتركة عند الجميع.
وكمثال .. دعنا نستخدم الفئة CSample التي أنشأناها في الدروس السابقة كفئة أساس تجمع الخصائص والطرق العامة والمشتركة لجميع الموظفين.
2- يمكننا الآن إنشاء عدة فئات أخرى تستخدم جميعها سمات الفئة الأساسية CSample .. قم بإنشاء فئة جديدة -(وحدة نمطية للفئة, وليست وحدة نمطية قياسية)- لكي تخدم كفئة مشتقة, واحفظها باسم CHourEmployee حيث ستمثل هذه الفئة مجموعة الموظفين الذين يتقاضون أجرهم بالساعة, ثم انسخ الكود التالي وألصقه فيها:

    Option Compare Database
Option Explicit

Private mHoursWorked As Double
Private mEmp As CSample

Private Sub Class_Initialize()
    Set mEmp = New CSample
End Sub

Private Sub Class_Terminate()
    Set mEmp = Nothing
End Sub

Public Property Get HoursWorked() As Double
    HoursWorked = mHoursWorked
End Property

Public Property Let HoursWorked(vNewValue As Double)
    mHoursWorked = vNewValue
End Property

Public Property Get Name() As String
    Name = mEmp.Name
End Property

Public Property Let Name(ByVal vNewValue As String)
    mEmp.Name = vNewValue
End Property

Public Property Get Salary() As Currency
    Salary = mEmp.Salary
End Property

Public Property Let Salary(ByVal vNewValue As Currency)
    mEmp.Salary = vNewValue
End Property

Public Function Tax() As Currency
    Tax = mEmp.Tax * 2
End Function
لاحظ ما يلي:

1- أن كود الفئة السابق يحتوي على كائن فئة أساس CSample , ومن أجل ذلك فقد احتجنا إلى حدث Initialize في الفئة المشتقة بهدف إنشاء مثيل جديد للفئة الأساس CSample .. فعندما يتم إنشاء مثيل جديد للفئة CHourEmployee سيتم قدح الحدث Initialize وتنفيذه بشكل آلي وبالتالي إنشاء مثيل جديد للفئة الأم CSample , وعندما يصبح المتغير الذي يمثل الفئة المشتقة CHourEmployee خارج المجال, سيتم تنفيذ الإجراء أو الحدث Terminate و بالتالي تدمير مثيل الفئة الأساس.
2- ترتبط الفئة المشتقة CHourEmployee مع الفئة الأم بشكل رخو, مما يعني أنك إذا أردت استخدام خصائص وطرق الفئة الأم في الفئة المشتقة, فستكون مضطراً إلى إضافة هذه الخصائص والطرق إلى الفئة المشتقة, وهذا هو السبب الذي دعا إلى قول القائلين: "إن الـ vba لا تدعم الوراثة المباشرة, ولا تعدد الأشكال الصريح", ولكن إذا نظرت إلى الدالة Tax فسترى كيف أنه يمكن لها أن تضيف مزايا حسابية تختلف عما كان موجوداً في الفئة الأساس.

اختبار الفئة المشتقة:

1- نقوم بإضافة فئة جديدة (وحدة نمطية للفئة) لتجمع عدد ساعات عمل كل شخص, ونحفظها باسم CWorkLogs , ونضع فيها الكود التالي:

كود:
  
Option Compare Database
Option Explicit

Private mColLogs As Collection

Public Property Get WorkLogs() As Collection
    Set WorkLogs = mColLogs
End Property

Public Property Set WorkLogs(lWorkLogs As Collection)
    Set mColLogs = lWorkLogs
End Property

Function TotalHoursWorked(strPersonName As String) As Double
    On Error GoTo Handle_Errors
    Dim wlog As CHourEmployee
    Dim doubleTotal As Double
    doubleTotal = 0
    For Each wlog In Me.WorkLogs
        If strPersonName = wlog.Name Then
            doubleTotal = doubleTotal + wlog.HoursWorked
        End If
    Next wlog

Exit_Here:
    TotalHoursWorked = doubleTotal
    Exit Function

Handle_Errors:
        'You will probably want to catch the error that will '
        'occur if WorkLogs has not been set '
        Resume Exit_Here
End Function
قم بإنشاء وحدة نمطية قياسية, وضع فيها الكود التالي:

كود:
     Option Compare Database
Option Explicit

Function PopulateHours() As Collection
    Dim collogs As Collection
    Dim helog As CHourEmployee
    'Put some data in the collection'
    Set collogs = New Collection

    Set helog = New CHourEmployee
    helog.Name = "Aref Hassan"
    helog.HoursWorked = 4.5
    collogs.Add helog

    Set helog = New CHourEmployee
    helog.Name = "Yosef Ahmed"
    helog.HoursWorked = 3
    collogs.Add helog

    Set helog = New CHourEmployee
    helog.Name = "Aref Hassan"
    helog.HoursWorked = 2.5
    collogs.Add helog

    Set PopulateHours = collogs
End Function

Sub GetHoursWorked()
    Dim pwl As CWorkLogs
    Set pwl = New CWorkLogs
    Set pwl.WorkLogs = PopulateHours()
    Debug.Print pwl.TotalHoursWorked("Aref Hassan")
    Set pwl = Nothing
End Sub
3- من النافذة السريعة قم بتنفيذ الإجراء GetHoursWorked وقل لي -بربك- ما الناتج؟.
في الموضوع القادم سنتطرق -إن شاء الله- إلى فئة المجموعة بتوسع و تفصيل أكثر مما ورد هنا, فإلى الملتقى.
فئات المجموعة Collection Classes - الجزء الأول
هذا هو الموضوع الثاني عشر في سلسلة البحث المسمى: إنشاء الفئات والبرمجة الموجهة للكائنات
عرفنا في الموضوع السابق أنه لتشكيل هرم من الفئات نحتاج إلى أن نصرح عن فئة -(وتدعى الفئة الأم أو الأساس)- داخل فئة أخرى -(وتدعى الفئة المشتقة)- وأنه يمكن لنا أن ننشئ العديد من الفئات المشتقة من نفس الفئة الأم, وأن العلاقة بين الفئتين -بالرغم من ضعفها- هي علاقة واحد إلى واحد, كما عرفنا أنه يمكن للفئة المشتقة أن تكون أماً لفئات أخرى بحيث يمكن للفئات المشتقة أن ترث أعضاء أمهاتها وجداتها على التعصيب أو النسب. واليوم نشرع في الحديث عن النوع الثاني من العلاقات, وهو علاقة واحد إلى متعدد, وبسبب أهمية هذا الموضوع, يجب أن نشرح لكم أولاً عن كائن المجموعة والذي هو حجر الزاوية في فهم العلاقة واحد إلى كثير, وبالتالي فهم فئة المجموعة.

كائن المجموعة Collection object سفينة نوح في البرمجة:

يحتوي الفيجوال بيسيك للتطبيقات على كائن يدعى Collection يمكن لنا أن نشبهه بسفينة نوح, من حيث قابليته أن يحمل في بطنه من كل زوجين اثنين, أنواعاً متباينة ومختلفة من الكائنات والبيانات دون تذمر ولا تأفف............
المجموعة Collection -من الكائنات التي يتم إنشاؤها بالكود- عبارة عن مجموعة كائنات يمكنك استخدامها بطريقة مشابهة للمجموعات المبنية في الاكسس مثل مجموعة الـ Forms والتي تعرفونها جيداً وسبق لكم التعامل معها ولا شك. سنقوم أولاً بدراسة المثال التالي الذي يعرف مجموعة تستخدم لحمل كائنات أو مثيلات للفئة CSample التي سبق شرحها:

كود:
Public Sub TestCollection()
    Dim colEmps As New Collection
    Dim Emp As CSample
    Set Emp = New CSample
    Emp.Name = "Aref Hassan"
    Emp.Salary = 350
    colEmps.Add Emp, Emp.Name
    Set Emp = Nothing
    
    Set Emp = New CSample
    Emp.Name = "Ahmed Al-Harbi"
    Emp.Salary = 5000
    colEmps.Add Emp, Emp.Name
    Set Emp = Nothing
    
    Set Emp = New CSample
    Emp.Name = "Yosef Ahmed"
    Emp.Salary = 4000
    colEmps.Add Emp, Emp.Name
    Set Emp = Nothing
    
    Set Emp = New CSample
    Emp.Name = "Naser Masarwa"
    Emp.Salary = 4500
    colEmps.Add Emp, Emp.Name
    Set Emp = Nothing
    
    For Each Emp In colEmps
        Debug.Print Emp.Name, Emp.Salary, Emp.Tax
    Next
    Debug.Print colEmps.Count
    Set Emp = colEmps("Aref Hassan")
    Debug.Print Emp.Name
    Debug.Print colEmps("Ahmed Al-Harbi").Name
    Debug.Print colEmps(3).Name
    Set colEmps = Nothing
    Set Emp = Nothing
End Sub
لاحظ ما يلي:

1- أنه تم استخدام المتغير Emp عدة مرات لإنشاء الكائنات(مثيلات الفئة), و بالرغم من أنه قد تم إعداده إلى Nothing -كما في الأسطر الملونة بالأحمر- إلا أن الكائن Emp لم يتم تدميره, والدليل على ذلك أن الدوارة قد طبعت لنا جميع أسماء الكائنات, والسبب في ذلك أنه عندما تم إنشاء الكائن Emp, قمنا بإضافته إلى المجموعة والتي أصبحت عندها مسؤولة عن إدارة الكائن. (عندما يتم إعداد المجموعة إلى Nothing سيتم تدمير الكائنات التي تحتويها).
2- أنه بالرغم من إعداد المجموعة إلى Nothing في السطر قبل الأخير, إلا أننا قمنا في السطر الأخير بإعداد المتغير Emp إلى Nothing فلماذا؟!! .. الجواب هو أن السبب يعود إلى أننا أنشأنا كائناً أو بالأحرى مرجعاً جديداً يشير إلى الفئة وذلك في السطر الملون بالأزرق في الكود أعلاه, فأصبح هذا الكائن خارج المجموعة, وبالتالي فتدمير المجموعة بما تحويه من كائنات لم يدمر هذا الكائن لأنه يقع خارج المجموعة, لذلك احتجنا إلى إعداد المتغير Emp إلى Nothing في هذه الحالة فقط التي وضعناها للتوضيح لتعرفوا كيف يتم تدمير الكائنات, وربما يكون لنا موضوع مستقل حول ذلك إن طال بنا العمر.
3- يحتوي كائن المجموعة Collection على عدة طرق مبنية منها: الطريقة Count و الطريقة Add كما وردتا أعلاه, وتسمح لك الطريقة Add بتعيين الوسيطتين Before و After بشكل اختياري لكي تقوم بوضع الكائن في الموضع الذي تراه مناسباً قبل أو بعد أي كائن سابق في المجموعة, ولمزيد من التفاصيل حول هذا الموضوع أنصحك بمراجعة تعليمات الاكسس.
4- العنصر الأول في المجموعة يأخذ الرقم1 وليس الرقم صفر كما في مجموعات اكسس المبنية, لذلك عندما طلبنا من الكود السابق أن يعطينا اسم العنصر رقم3 في المجموعة, فقد أعطانا اسم أستاذنا القدير يوسف أحمد.
5- عندما نقوم بالإشارة إلى إعضاء الفئة عبر المجموعة مثال: colEmps(3).Name فأننا لا نستطيع الاستفادة من ميزة Intellisense وذلك بسبب أن هذا النوع من المجموعات يمكن أن يحمل أنواعاً مختلفة من الكائنات, لذلك لا تستطيع بيئة التطوير أن تعرف بالضبط ما هي الخصائص التي سيتم إعطاءها للكائن.

الفرق بين المجموعات المبنية في الاكسس و المجموعة التي ننشئها بالكود:

1- الفرق الأول هو أن المجموعات المبنية في الاكسس مثل مجموعة TableDefs تحمل نوعاً واحداً فقط من الكائنات, بينما المجموعة التي ننشئها في الـ vba مثل المجموعة في الكود السابق يمكن لها أن تحمل أنواعاً مختلفة من الكائنات أو الفئات ضمن نفس المجموعة, وهذا ما يفسر محدودية خاصية Intellisense .
2- الفرق الثاني هو أن مجموعات الـ vba ذات أساس 1 بينما مجموعات الاكسس المبنية ذات أساس صفر.

في الموضوع القادم -إن شاء الله- سنعرف كيف يعمل كائن المجموعة من داخل فئة المجموعة, بالإضافة إلى المزيد من المعلومات عن فئة المجموعة, فإلى الملتقى.
فئات المجموعة Collection Classes الجزء الثاني
هذا هو الموضوع الثالث عشر في سلسلة البحث المسمى: إنشاء الفئات والبرمجة الموجهة للكائنات
تحدثنا في الموضوع السابق عن كائن vba يدعى collection وبأن هذا الكائن يمكن له أن يحمل أي نوع من البيانات حتى ولو كانت متباينة, وقلنا -أيضاً- بأن هذا الكائن يلعب دوراً محورياً في إنشاء فئة المجموعة والتي تعتبر الركيزة الأساسية لعلاقة واحد إلى متعدد بين الفئات, واليوم سنتحدث عن كيفية إنشائها, وماهي مزاياها, وعيوبها, وكيفية التغلب على هذه العيوب.

إنشاء فئة مجموعة Collection Class :
يدعم كائن المجموعة collection عدداً محدوداً فقط من الطرق وهي: Item و Add و Count و Remove وسترغب عادة في إضافة المزيد من الطرق إلى المجموعة, وستحتاج لعمل ذلك إلى تعريف فئة مجموعة خاصة بك, وسنطلق عليها هنا اسم CSampleCollection .
خطوات إنشاء وتعريف فئة مجموعة, هي نفس خطوات إنشاء و تعريف أي وحدة نمطية للفئة, وستحتوي فئة المجموعة على كائن مجموعة Collection -كما عرفنا في الموضوع السابق- لذلك سنقوم بتعريف متغير خاص (داخلي) في فئة المجموعة يدعى mcolSamples , وكما عرفنا سابقاً أيضاً بأن الفئات يمكن أن يكون لها حدثان خاصان بها وهما: Initialize و Terminate يتم قدحهما عند إنشاء مثيل الفئة وعند إنهاء ذلك المثيل على التوالي. لا تحتاج الفئة البسيطة التي أنشأناها سابقاً والمسماة CSample لأيٍّ من هذين الحدثين, لكن الفئة الجديدة CSampleCollection تحتاج لإنشاء كائن مجموعة Collection وذلك في الحدث Initialize , ثم تحتاج بعد ذلك إلى إزالة جميع الكائنات من المجموعة عندما يتم إنهاؤها وذلك في الحدث Terminate كما هو موضح في الكود التالي:
قم بنسخ الكود التالي, ثم الصقه في وحدة نمطية للفئة, واحفظها باسم CSampleCollection :

كود:
Private mcolSamples As Collection

Private Sub Class_Initialize()
    Set mcolSamples = New Collection
End Sub

Private Sub Class_Terminate()
    Dim aSample As CSample
    For Each aSample In mcolSamples
        mcolSamples.Remove CStr(aSample.Name)
    Next
    Set mcolSamples = Nothing
End Sub
ولكي تعمل هذه الفئة كفئة مجموعة, فأنت بحاجة إلى أن تحتوي الفئة على جميع الطرق الموجودة في كائن المجموعة من أجل تعداد, وإضافة, وإزالة العناصر من الفئة, لذلك يجب إضافة هذه الطرق إلى فئة المجموعة الخاصة بك. قم -كذلك- بإضافة الطريقة Item والتي تعتبر أيضاً من السمات الأساسية للفئة. انسخ الكود التالي, ثم أضفه إلى أسفل الكود السابق:

كود:
Public Property Get count() As Long
    count = mcolSamples.count
End Property

Sub Add(Sample As CSample)
    mcolSamples.Add Sample, Sample.Name
End Sub

Sub Remove(sName As Variant)
    mcolSamples.Remove CStr(sName)
End Sub

Function Item(sName As Variant) As CSample
    Set Item = mcolSamples(sName)
End Function
المشكلة مع فئات المجموعة:

حالما تبدأ في إنشاء وتعريف فئة مجموعة خاصة بك, ستجد أن عدداً من الخصائص المعروفة لكائنات المجموعة Collection لم تعد تعمل. فمثلاً لا تستطيع استخدام الدوارة For Each أو فهرسة المجموعة باستخدام اسم مفتاح. 
يستخدم الإجراء التالي -في وحدة نمطية قياسية- لاختبار فئة المجموعة التي أنشأناها في الكود السابق, ولتوضيح المشكلة فقد تم تلوين بعض سطور الكود بالأحمر للإشارة إلى أن هذه الأسطر لن تعمل في فئة المجموعة الخاصة بنا, وستكون -تلك الأسطر- سبباً في ظهور رسالة خطأ أثناء التشغيل برقم 438 والتي تقول: "الكائن لا يدعم هذه الطريقة أو الخاصية", (يمكنك -بعد التعرف على المشكلة- تعليق أو حذف الأسطر الحمراء إن شئت).

كود:
Sub CSampleCollectionTest()
    Dim AllSamples As New CSampleCollection
    Dim Sample As CSample
    Dim lngCount As Long
    Set Sample = New CSample
    Sample.Name = "Ahmed"
    Sample.Salary = 5000
    AllSamples.Add Sample
    Set Sample = Nothing
    
    Set Sample = New CSample
    Sample.Name = "Naser"
    Sample.Salary = 2000
    AllSamples.Add Sample
    Set Sample = Nothing
    
    Set Sample = New CSample
    Sample.Name = "Aref"
    Sample.Salary = 400
    AllSamples.Add Sample
    Set Sample = Nothing

    Set Sample = New CSample
    Sample.Name = "Yosef"
    Sample.Salary = 4000
    AllSamples.Add Sample
    Set Sample = Nothing
    
    'Next lines will not work
    For Each Sample In AllSamples
        Debug.Print Sample.Name, Sample.Salary, Sample.Tax
    Next
    
    'But next lines will work
    For lngCount = 1 To AllSamples.count
        Set Sample = AllSamples.Item(lngCount)
        Debug.Print Sample.Name, Sample.Salary, Sample.Tax
    Next
    Set Sample = Nothing
    
    'Next line will not work
    Set Sample = AllSamples("Aref")
    
    'But next line will work
    Set Sample = AllSamples.Item(3)
    Debug.Print AllSamples.Item(1).Name
    Debug.Print Sample.Name
    Set Sample = Nothing
    Set AllSamples = Nothing
    
End Sub
طرق حل المشكلة:

توجد تقنيتان للتحايل على هذه المشكلة التي تسبب عدم القدرة على الإشارة إلى مجموعة الفئة عبر استخدام مفتاح الأسماء. يتم في التقنية الأولى إضافة دالة AllItems إلى فئة المجموعة الخاصة بك, ويتم في التقنية الثانية تصدير فئة المجموعة إلى ملف نصي, وتحريرها في الملف النصي, ثم إعادة استيراد الفئة بعد ذلك. وإليكم شرح الطريقتين.

1- التقنية الأولى إضافة دالة AllItems إلى الفئة:

نقوم بإضافة الطريقة التالية إلى الوحدة النمطية لفئة المجموعة:

كود:
Public Function AllItems() As Collection
    Set AllItems = mcolSamples
End Function
ولكي يعمل الكود السابق -الذي كان يسبب الخطأ- بدون أي مشاكل, سنقوم بالتعديل عليه لتضمين الدالة الجديدة بحيث يصبح الكود كما يلي:

كود:
    For Each Sample In AllSamples.AllItems
        Debug.Print Sample.Name, Sample.Salary, Sample.Tax
    Next
    '
    Set Sample = AllSamples.AllItems("Aref")
وسنجد أن الكود يعمل بشكل ممتاز, وهذا حلٌّ مرضٍ, لأننا فقط قمنا بإدراج المشير AllItems عندما استخدمنا المجموعة مع مفتاح الكائن.

2- التقنية الثانية تصدير وإعادة استيراد الفئة:

السبب الحقيقي في عدم قدرتنا على الإشارة إلى فئة المجموعة باستخدام الصيغة المعروفة القياسية, هو أن فئات الفيجوال بيسيك للتطبيقات -وأعتقد أن فئات الفيجوال بيسيك6 كذلك- لا تسمح بأن يتم تعديل بعض السمات الخاصة للفئات, وهي السمات المطلوبة لدعم الصيغة المعروفة, لذلك نلجأ لهذه التقنيات للالتفاف حول تلك السمات كما في التقنية الأولى, أو لتعديل تلك السمات كما في هذه التقنية التي سأوضحها الآن.
في نافذة المشروع, قم بالنقر بالزر الأيمن على الوحدة النمطية لفئة المجموعة, ومن القائمة المنسدلة اختر التصدير إلى ملف نصي, ثم افتح الملف النصي المفكرة وستشاهد البيانات التالية:

كود:
VERSION 1.0 CLASS
BEGIN
  MultiUse = -1  'True
End
Attribute VB_Name = "CSampleCollection"
Attribute VB_GlobalNameSpace = False
Attribute VB_Creatable = False
Attribute VB_PredeclaredId = False
Attribute VB_Exposed = False
Option Compare Database
Option Explicit

Private mcolSamples As Collection

Private Sub Class_Initialize()
    Set mcolSamples = New Collection
End Sub

Private Sub Class_Terminate()
    Dim aSample As CSample
    For Each aSample In mcolSamples
        mcolSamples.Remove CStr(aSample.Name)
    Next
    Set mcolSamples = Nothing
End Sub

Public Property Get count() As Long
    count = mcolSamples.count
End Property

Sub Add(Sample As CSample)
    mcolSamples.Add Sample, Sample.Name
End Sub

Sub Remove(sName As Variant)
    mcolSamples.Remove CStr(sName)
End Sub

Function Item(sName As Variant) As CSample
    Set Item = mcolSamples(sName)
End Function
لايتم عرض هذه السمات في بيئة الفيجوال بيسيك للتطبيقات. أليس كذلك؟
لكي نتمكن من حل المشكلة السابقة, نحتاج لأن نجعل أحد عناصر فئة المجموعة هو العنصر الافتراضي للمجموعة, وهذا العنصر تحديداً هو الطريقة Item , لذلك نحن بحاجة إلى تعديل الطريقة في المفكرة بإضافة تعريف السمة التالي:
كود:
Function Item(sName As Variant) As CSample
    Attribute Item.VB_UserMemId = 0
    Set Item = mcolSamples(sName)
End Function
ومن أجل دعم التعداد في دوارة For Each نحتاج أيضاً لإضافة طريقة NewEnum إلى المفكرة كالتالي:

كود:
Function NewEnum() As IUnknown
    Attribute NewEnum.VB_UserMemId = -4
    Set NewEnum = mcolSamples.[_NewEnum]
End Function
بعد حفظ التغييرات في المفكرة, قم بإعادة استيراد الفئة إلى المشروع.
إذا نظرت إلى محرر الفيجوال بيسيك فلن تستطيع رؤية السمات التي قمت بتعديلها وإضافتها إلى الطريقة Item لأنها تبقى مخفية.
الآن يمكن للكود ذي الأسطر الحمراء في الأعلى أن يعمل دون أن تظهر أي رسائل خطأ أثناء التشغيل.

في الموضوع القادم -إن شاء الله- سنرى كيفية إنشاء الأحداث الخاصة بالمستخدم, فإلى الملتقى.
استخدام الفئات Events & Classes والأحداث 
هذا هو الموضوع الرابع عشر في سلسلة البحث المسمى: إنشاء الفئات والبرمجة الموجهة للكائنات
عرفنا في الموضوع السابق كيفية إنشاء فئة مجموعة, وأرجو -اعتماداً على ما سبق- أن تجربوا عمل فئة مجموعة بأنفسكم, لأنكم ستكتشفون الكثير مما قيل, والكثير مما لم يقل خطأً أو سهواً -وجل من لا يسهو- , وإن واجهتكم صعوبة أو سؤال, فسنناقشه معاً إن شاء الله. 
واليوم سنتناول كيفية إنشاء الأحداث الخاصة المعرفة من قبل المستخدم.
لماذا نحتاج إلى إنشاء الأحداث في الفئات:
1- لإرسال إشارات وتنبيهات إلى المستخدم ببدء عملية معينة أو انتهائها.
2- لإعطاء المستخدم المقدرة على إلغاء بعض العمليات عند حدوث أفعال معينة, أو تحقق شروط محددة.

كيفية إنشاء الأحداث في الفئات:

يمكن لمبرمج الفئات أن ينشئ في الفئات أحداثاً خاصة به ويتم قدحها في الزمان والمكان اللذين يراهما مناسبين في سياق الكود البرمجي للفئة. فعندما يشعر مبرمج الفئة بالحاجة إلى عرض أحداث معينة على مستخدم الفئة, فإنه يقوم بالإعلان عن هذه الأحداث في قسم التصريحات العامة للفئة, وهناك قواعد عامة يجب معرفتها والالتزام بها عند إنشاء الأحداث وهي:

1- لا يمكن إنشاء الأحداث إلا في الوحدات النمطية للفئات, أي أنه لا يمكن لنا إنشاء حدث في وحدة نمطية قياسية.
2- إذا أنشأنا حدثاً في فئة ما, فيجب أن يكون هذا الحدث من النوع Public ولا يمكن أبداً أن يكون من النوع Private .
3- كما أنه لا يمكن إنشاء الأحداث إلا في الفئات, فكذلك لا يمكن أن تستعمل وتستخدم هذه الأحداث سوى في الوحدات النمطية للفئات فقط, سواءاً أكانت وحدة نمطية لنموذج, أو لتقرير, أو لفئة تم إنشاؤها من قبل المبرمج.

4- الصيغة العامة للإعلان عن حدث, والتي يتم وضعها في قسم التصريحات العامة للفئة هي كما يلي:
كود:
Public Event EventName(Parameters, ByRef Cancel As Boolean)
حيث Public Event عبارة عن الكلمة المفتاحية للإعلان عن الحدث, و EventName هو اسم الحدث الذي تريد إنشاؤه, والـ Parameters هي معاملات الحدث -إن وجدت- مثل intI As Integer وهي اختيارية ويمكنك تركها فارغة, وأخيراً Cancel وهو اختياري -أيضاً- ويمكنك تركه فارغاً ووظيفته هو السماح لمستخدم الحدث بأن يلغي عملية معلقة, أو أن يلغي عملية قيد التنفيذ.

5- نحتاج إلى الكلمة المفتاحية Public Event لكل حدث تتم إضافته إلى الفئة, كما نحتاج إلى الكلمة المفتاحية RaiseEvent في المكان المناسب من كود الفئة حيث نريد أن يقدح الحدث ويظهر إلى العلن. ويعتمد هذا الموضع -بالكلية- على سياق تنفيذ الكود, وعلى ماهية الفعل أو الشرط الذي يعرّفه الحدث.

6- مما سبق نعلم ونستنتج أن: هناك فئتان, فئة يتم فيها إنشاء وبناء الحدث, وفئة أخرى يتم فيها استخدام واستهلاك الحدث وتسمى اصطلاحاً بالفئة المستقبلة أو المستهلكة للحدث.

الكلمة المفتاحية WithEvents :

عندما نريد استخدام الفئة مع أحداثها, -أو استخدام أي كائن مع أحداثه, كالنموذج أو عنصر التحكم- فإننا نقوم بالإعلان -في الفئة المستهلكة- عن متغير من نوع الفئة المنشئة للحدث, أو من نوع النموذج, أو عنصر التحكم مع استخدام الكلمة المفتاحية WithEvents لكي نتمكن من العمل مع أحداث ذلك المتغير, والصيغة العامة لذلك كالتالي:

كود:
Dim WithEvents variableName As ObjectName
وبمجرد الإعلان عن المتغير بالطريقة السابقة, سنجد أنه قد تم إدراجه في مربع التحرير والسرد الموجود في أعلى وأيسر محرر الفيجوال بيسيك (يحتوي عادة على كلمة General ) قم بنقر السهم الصغير المتجه إلى أسفل لتظهر قائمة منسدلة تحتوي على المتغير الذي أعلنت عنه للتو. قم باختيار المتغير من تلك القائمة وستجد أن القائمة الأخرى في الجهة اليمنى تحتوي على كل أحداث المتغير الذي أعلنت عنه, وبمجرد النقر على أي حدث, فسيتم إدراجه في الوحدة النمطية للفئة بشكل قالب جاهز بما يشبه إضافة الأحداث إلى نمطية النماذج.
مثال عملي:
سنقوم بإضافة بعض الأحداث إلى الفئة CSampleCollection التي أنشأناها في الموضوع السابق, بحيث نعطي مستخدم الفئة الحرية في أن يمنع من يشاء من أن تتم إضافته إلى فئة المجموعة, وسنقوم بالتعديل على كود الفئة CSampleCollection ليصبح كالتالي:
ود:
Option Compare Database
Option Explicit

Private mcolSamples As Collection
Public Event BeforeAdd(Cancel As Integer)
Public Event AfterAdd()
Public Event InValidAddition()
Private mCancel As Integer

Private Sub Class_Initialize()
    Set mcolSamples = New Collection
End Sub

Private Sub Class_Terminate()
    Dim aSample As CSample
    For Each aSample In mcolSamples
        mcolSamples.Remove CStr(aSample.Name)
    Next
    Set mcolSamples = Nothing
End Sub

Public Property Get count() As Long
    count = mcolSamples.count
End Property

Sub Add(Sample As CSample)
    RaiseEvent BeforeAdd(mCancel)
    If Not mCancel Then
        mcolSamples.Add Sample, Sample.Name
        RaiseEvent AfterAdd
    Else
        RaiseEvent InValidAddition
    End If
End Sub

Sub Remove(sName As Variant)
    mcolSamples.Remove CStr(sName)
End Sub

Function Item(sName As Variant) As CSample
    Set Item = mcolSamples(sName)
End Function

Function NewEnum() As IUnknown
        Set NewEnum = mcolSamples.[_NewEnum]
End Function
بعد ذلك -ولكي نختبر التعديلات التي أجريناها على الفئة- سنحتاج إلى نموذج وسنضيف إليه ثلاث مربعات نص, نجعل اسم الأول sName لإدخال اسم الموظف, ونجعل اسم مربع النص الثاني sDOB لإدخال تاريخ الميلاد, ونجعل اسم مربع النص الثالث sSalary لإدخال المرتب. كما سنضيف إليه زر أمر باسم cdAdd لإضافة البيانات التي أدخلناها في مربعات النص إلى فئة المجموعة, ونضع الكود التالي في الوحدة النمطية للنموذج:

كود:
Option Compare Database
Option Explicit

Dim WithEvents myObjCol As CSampleCollection
Dim S As New CSample


Private Sub cdAdd_Click()
    If Nz(Me.sName, "") = "" Or (Not IsDate(Me.sDOB)) Or _
        Nz(Me.sSalary, 0) = 0 Then Exit Sub
    LoadSample
    myObjCol.Add S
    Set S = Nothing
End Sub

Private Sub Form_Load()
    Set myObjCol = New CSampleCollection
End Sub

Private Sub myObjCol_AfterAdd()
    MsgBox "Successfully Added to the class collection", _
        vbOKOnly, "Event Tester"
    Me.sName = ""
    Me.sDOB = ""
    Me.sSalary = 0
End Sub

Private Sub myObjCol_BeforeAdd(Cancel As Integer)
    If Me.sName = "Aref Hassan" Then
        Cancel = True
    Else
        Cancel = False
    End If
End Sub

Sub LoadSample()
    S.Name = Me.sName
    S.DOB = Me.sDOB
    S.Salary = Me.sSalary
End Sub

Private Sub myObjCol_InValidAddition()
    MsgBox "I'm sorry, this person cann't be add to data." & vbNewLine & _
    "Only authorised people can be accepted.", vbCritical, "Invalid person"
    Me.sName = ""
    Me.sDOB = ""
    Me.sSalary = 0
End Sub
قم بحفظ النموذج بأي اسم تريد, ثم قم بإدخال بيانات وهمية في مربعات النص وانقر زر إضافة وستجد أن البيانات يتم إضافتها بشكل صحيح إن كانت من نوع البيانات الصحيحة, وستحصل على رسالة - بسبب الحدث AfterAdd - تفيد بنجاح الإضافة إلى فئة المجموعة, أما إذا قمت بإدخال اسم Aref Hassan في خانة الاسم مع إدخال تاريخ الميلاد والمرتب, فسيتسبب ذلك في نشؤ الحدث InValidAddition كما في الأكواد السابقة وستحصل على رسالة مختلفة.
أتمنى أن أكون قد وفقت في إيصال الفكرة إلى أذهانكم, وأن تكونوا قد استفدتم من هذا الموضوع ومن المواضيع السابقة.
في الموضوع القادم -إن شاء الله- سنتطرق إلى الفئات المجردة Abstract Classes والفئات المنفذة Implementation Classes فإلى الملتقى.
الفئة المجردة الهجينة Hybrid Abstract Class

هذا هو الموضوع السادس عشر في سلسلة البحث المسمى: إنشاء الفئات والبرمجة الموجهة للكائنات
عرفنا في الموضوع السابق أن الفئات المجردة عبارة عن قوالب لصياغة فئات أخرى تدعى بالفئات المنفذة أو الفئات المشتقة, وأنه يطلق على الفئات المجردة تسمية أخرى وهي فئات الواجهة, وتوحي لنا هذه التسمية بالوراثة, وتعدد الأشكال, لكننا رأينا في الموضوع السابق, كما رأينا في موضوع التسلسل الهرمي للفئات, بأن الـ vba لا تدعم تعريفات تعدد الأشكال ولا الوراثة كما هي معروفة في اللغات الأخرى الداعمة للبرمجة الموجهة للكائنات, وبالرغم من ذلك فالـ vba لديها بعض السمات الذكية كاستخدام WithEvents التي تذهب مذاهب بعيدة ما وراء تعريفات البرمجة الموجهة للكائنات.
ماهي الفئة المجردة الهجينة:
هي فئة تجمع ما بين صفات الفئات غير المجردة, وصفات الفئات المجردة, فهي فئة مجردة وغير مجردة هجينة .. كيف ذلك؟.
إن مصطلح الفئة المجردة يعني أنك لا تنشئ مثيلات Instances عن الفئة, ولكن بسبب أن الفئة المجردة في الفيجوال بيسيك ليست بدعاً من الفئات, لذلك يمكن لك أن تنشئ مثيلاً Instance عن الفئة المجردة داخل الفئة المنفذة, وبذلك تكون قد أضفت مرجعاً يشير إلى مثيل للفئة المجردة باستخدام شكل من التغليف حول الفئة المجردة, وقد يكون هذا شكلاً فقيراً من أشكال الوراثة, لكنه يقود إلى بعض الاحتمالات المثيرة حيث من الممكن أن نخلط ونطابق السمات من فئات مجردة وغير مجردة عند تنفيذ واجهة ما.
كيفية إنشاء فئة هجينة:
بالرغم من أن الإعلان عن المتغيرات وكتابة الأكواد مسموح بهما في الفئات المجردة, وبالرغم من أن ذلك يعد عند البعض عملاً غير مستحسن, لأنه لا يمكنك استخدام هذه المتغيرات أو أي كود تتم كتابته في الفئة المجردة, (لأن الفئة المجردة -كما قلنا- عبارة عن قالب), إلا أنه عند حدوث ذلك فإنك تكون قد أنشأت فئة هجينة مجردة وغير مجردة.
لتوضيح كيفية عمل هذا, سنقوم بإنشاء الفئة الهجينة ISample2 وهي فئة ستخدم كفئة قاعدة أساس Base Class -(راجع موضوع التسلسل الهرمي للفئات)- وستخدم أيضاً كفئة مجردة أو فئة واجهة.
1- قم بنسخ الفئة ISample التي أنشأناها في الموضوع السابق, واحفظ النسخة باسم ISample2 .
2- قم بإضافة متغير خاص إلى قسم التصريحات العامة للفئة ISample2 باسم mNationality من نوع سلسلة نص.
3- قم بإضافة خاصية للفئة ISample2 باسم Nationality من نوع سلسلة نصية, وقم بتعيين قيمة الخاصية إلى المتغير الخاص mNationality .
أصبحت -الآن- الفئة ISample2 فئة هجينة في أن جزءاً منها ماديٌ, والجزء الآخر منها مجرد, وأصبح كودها يشبه التالي:
كود:
Option Compare Database
Option Explicit

Private mNationality As String


Public Property Get Name() As String

End Property

Public Property Let Name(ByVal vNewValue As String)
    
End Property

Public Property Get DOB() As Date
    
End Property

Public Property Let DOB(ByVal vNewValue As Date)
        
End Property

Public Property Get Age() As Byte
    
End Property

Public Property Get Salary() As Currency
    
End Property

Public Property Let Salary(ByVal vNewValue As Currency)
    
End Property

Public Function Tax() As Currency
    
End Function

Public Property Get Nationality() As String
    Nationality = mNationality
End Property

Public Property Let Nationality(ByVal vNewValue As String)
    mNationality = vNewValue
End Property
والآن سنقوم بإنشاء فئة منفذة ولتكن باسم CRegularEmp2 وسنحتاج فيها إلى التعليمة Implements لتنفيذ الفئة الهجينة -كما رأينا ذلك في الموضوع السابق- , وسنحتاج في الفئة المنفذة أيضاً إلى إنشاء مثيل عن الفئة الهجينة على أنها فئة أساس (ونكون بذلك قد قمنا بتنفيذ وتغليف الفئة الهجينة في نفس الوقت). بحيث يكون كود الفئة المنفذة CRegularEmp2 كما يلي:

كود:
Option Compare Database
Option Explicit

Implements ISample2
Private mHybClass As ISample2
Private mSalary As Currency
Private mSName As String
Private mDOB As Date

Private Sub Class_Initialize()
    Set mHybClass = New ISample2
End Sub

Private Sub Class_Terminate()
    Set mHybClass = Nothing
End Sub

Private Property Get ISample2_Age() As Byte
    ISample2_Age = DateDiff("yyyy", mDOB, Date)
End Property

Private Property Let ISample2_DOB(ByVal RHS As Date)
    mDOB = RHS
End Property

Private Property Get ISample2_DOB() As Date
    ISample2_DOB = mDOB
End Property

Private Property Let ISample2_Name(ByVal RHS As String)
    mSName = RHS
End Property

Private Property Get ISample2_Name() As String
    ISample2_Name = mSName
End Property

Private Property Let ISample2_Salary(ByVal RHS As Currency)
    mSalary = RHS
End Property

Private Property Get ISample2_Salary() As Currency
    ISample2_Salary = mSalary
End Property

Private Function ISample2_Tax() As Currency
    ISample2_Tax = mSalary * 0.1
End Function

Private Property Let ISample2_Nationality(ByVal RHS As String)
    mHybClass.Nationality = RHS
End Property

Private Property Get ISample2_Nationality() As String
    ISample2_Nationality = mHybClass.Nationality
End Property
لاحظ ما يلي:

- لأنك تقوم بتنفيذ الفئة الهجينة كفئة مجردة عبر الكلمة المفتاحية Implements فلا تزال بحاجة إلى الولوج إلى الخاصية Nationality كي تقوم بتعيين أو بقراءة قيمتها, لكن -تذكر- أنك تقوم باستدعاء الكود الموجود في الفئة الهجينة من داخل الفئة المنفذة, لذلك احتجنا إلى عمل مثيل mHybClass للفئة الهجينة في الحدث Initialize للفئة المنفذة, وإنهاؤه في الحدث Terminate كما تعلمنا في موضوع التسلسل الهرمي للفئات.
- تأمل -يا رعاك الله- كيف تم التصريح عن المتغيرات الخاصة لجميع الخصائص في الفئة المنفذة CRegularEmp2 ماعدا الخاصية Nationality فقد اختلفت طريقة إسناد القيم إليها .. أليس كذلك؟
ومن أجل تجربة الفئات السابقة قم بوضع هذا الكود في وحدة نمطية قياسية, ثم قم بتشغيله:

كود:
Sub testHybClass()
    Dim Emp As ISample2
    Set Emp = New CRegularEmp2
    Emp.Name = "Aref Hassan"
    Emp.DOB = #9/10/1965#
    Emp.Salary = 400
    Emp.Nationality = "Yemeni"
    Debug.Print Emp.Name, Emp.Age, Emp.Tax, Emp.Nationality
    Set Emp = Nothing
End Sub
'Output:
'testHybClass
'Aref Hassan    49            40           Yemeni
في الموضوع القادم -إن شاء الله- سنتطرق إلى معالجة الأخطاء في الفئات, فإلى الملتقى.
معالجة الأخطاء في الفئات
هذا هو الموضوع السابع عشر في سلسلة البحث المسمى: إنشاء الفئات والبرمجة الموجهة للكائنات
عرفنا في الموضوع السابق أن الفئات الهجينة تجمع بين سمات الفئات المجردة وغير المجردة, وأنه لكي نصل إلى واجهات الفئة الهجينة فأننا نستخدم الكلمة المفتاحية Implements في الفئة المنفذة مع إنشاء مثيل للفئة الهجينة في الفئة المنفذة, وبذلك تكون الفئة الهجينة قد جمعت الوراثة من طرفيها, وراثة الواجهات, والوراثة بالتفويض (للمزيد من المعلومات راجع المواضيع السابقة). واليوم سنتطرق إلى موضوع معالجة الأخطاء في الفئات.
تصيد الأخطاء Error Trapping :
تقوم كثير من البرامج في مختلف لغات البرمجة بتصيد ومعالجة الأخطاء, ومعظم هذه المعالجات تكون إما استجابة لأخطاء بحسب ظروف الحالة Status Condition , وإما حماية للبيانات من أخطاء المستخدم.
الأخطاء بحسب الظروف: عبارة عن أخطاء يتم توليدها خارج الفئة حسب الأكواد التي يكتبها المستخدم, وغالباً ما تكون معظم هذه الأخطاء غير متوقعة و غير مرحب بها, و ستجد بأنك لست - فقط- مضطراً لعمل مصيدة لهذه الأخطاء ومعالجتها, بل وتحتاج -أيضاً- إلى فهم كيفية العمل مع الأخطاء في الوحدات النمطية للفئات.

كيف يتعامل الفيجوال بيسيك للتطبيقات مع الأخطاء:

عند حدوث خطأ في إجراء لا يحتوي على معالجة للخطأ, يقوم الفيجوال بيسيك للتطبيقات بتمرير الخطأ إلى الإجراء التالي الأعلى في سلسلة الاستدعاءات, فإن لم يجد معالجاً للخطأ في الإجراء الأعلى, فإنه يستمر بتمرير الخطأ صعوداً حتى يصل إلى أعلى مستوى والذي يتم فيه إظهار رسالة الخطأ المعروفة "خطأ أثناء التشغيل" أو Run_time_error . فإذا لم تقم بعمل معالجة للأخطاء عند كتابتك للأكواد, فسيقوم الـ vba بمعالجتها, لكنك قد تكره هذه المعالجة عند قيامها بإعادة ضبط المتغيرات في برنامجك, ولا شك أن مستخدمي برنامجك سيشعرون -جراء ذلك أيضاً- بالامتعاض وخيبة الأمل.

كيفية معالجة الأخطاء في الفئات:

معالجة الأخطاء في الفئات هي نفس المعالجة في الأكواد الأخرى, فيما عدا أنه يوجد خيار لضبط إعدادات مصيدة الأخطاء وقت التشغيل خاص بالفئات, وذلك من خيارات محرر بيئة التطوير IDE , اختر التبويب عام General وستجد في الجهة اليمنى من النافذة ثلاثة خيارات لتصيد الخطأ Error Trapping كما في الصورة التالية:

تضبط هذه الإعدادات كيفية معالجة الاكسس للأخطاء, وإليكم شرح هذه الخيارات:
1- الإعداد الافتراضي وقت التشغيل هو Break on Unhandled Errors وفيه يقوم الاكسس بإظهار رسالة خطأ وقت التشغيل القياسية والمعروفة وذلك عند حدوث خطأ لم يقم المبرمج بعمل المعالجة اللازمة له, وهذا الخيار مرغوب به, لأنه يمكّن معالجات الخطأ التي تكتبها في إجراءاتك بأن تقوم بعملها.
2- الخيار Break on All Errors يجعل الاكسس يتجاهل جميع معالجات الخطأ التي لديك, ويظهر رسالة خطأ وقت التشغيل عند حدوث خطأ في أي وحدة نمطية في التطبيق سواءاً أكانت وحدة نمطية قياسية, أم وحدة نمطية للفئة.
3- أما الخيار Break on Class Module فيجعل الاكسس يتجاهل معالجات الخطأ في الوحدات النمطية للفئات فقط, ولا يتجاهل معالجات الخطأ في الوحدات النمطية القياسية.

التحكم في نصوص رسائل خطأ وقت التشغيل: 
إظهار رسائل أخطاء معرفة من قبل المبرمج Raising Custom_Defined Errors :

قد تكون رسائل الأخطاء أثناء التشغيل - أحياناً- غير مفهومة وليست واضحة للمستخدمين, لذلك يلجأ بعض المبرمجين إلى تغيير نصوص تلك الرسائل بنصوص تكون أكثر وضوحاً ولطفاً وخصوصية وقبولاً لمستخدمي برامجهم.
يوجد كائن في الـ vba يدعى الـ Err , وتوجد لدى هذا الكائن طريقة Method اسمها Raise والصيغة العامة لاستخدامها كالتالي:

كود:
Err.Raise Number, Source, Descreption, HelpFile, HelpContext
وللمزيد من المعلومات حول هذا الكائن أو معاملاته (وسائطه), اكتب في محرر الفيجوال بيسيك Err.Raise ثم ضع مؤشر الكتابة فوق الكلمة Raise وانقر بعد ذلك -من لوحة المفاتيح- الزر F1 .

ولمنع التضارب مع الأخطاء المبنية في الاكسس أو مع العناصر الأخرى يجب أن تضيف إلى الكود ثابتاً اسمه vbObjectError كالتالي مثلاً:
على سبيل المثال سنقوم بتعديل الدالة Tax والموجودة في الفئة Csample لكي يتم تنبيه المستخدم في حال لم يقم بإسناد قيمة إلى الخاصية Salary عند تشغيل الدالة:

كود:
Public Function Tax() As Currency
    On Error GoTo TaxErrHandler
    Const myContextId = 1010407
    Dim strErrDesc As String
    If Me.Salary <= 100 Then
        Tax = 0.1 * Me.Salary
    Else
        Tax = 0.12 * Me.Salary
    End If
TaxErrHandler:
    If Err.Number = 5 Then
        strErrDesc = "Sorry, you didn't assign any value to Salary. Plz assign it."
        Err.Raise vbObjectError + 5, TypeName(Me), strErrDesc
    Else
        Err.Raise Err.Number, Err.Source, Err.Description
    End If
End Function
لاحظ ما يلي:

- أننا أضفنا الثابت vbObjectError إلى رقم الخطأ لمنع تضارب رسالة الخطأ هذه مع رسالة خطأ وقت التشغيل رقم 5.
- الدالة TypeName تعيد نوع بيانات المتغير الممرر إليها, إلا إذا كان المتغير من نوع مثيل لفئة Instance فإنها تعيد اسم ذلك المثيل.
- أن TaxErrHandler لا تعتبر معالجة للخطأ حتى وإن كان الاسم يوحي بذلك, فكل ما قمنا بعمله ما هو إلا استبدال نص رسالة خطأ أثناء التشغيل فقط, وسيتم تمرير الخطأ إلى الكود المستدعي كما سنرى لاحقاً.
ومن أجل أن تشاهد النتيجة, قم بوضع الكود التالي في وحدة نمطية قياسية, ثم قم بتشغيله:

كود:
Sub testClassError()
    Dim obj As CSample
    Set obj = New CSample
    Debug.Print obj.Tax
    Set obj = Nothing
End Sub
لاحظ ما يلي:

- تعمدت في الكود السابق أن أخطئ في كتابة الكود لكي يظهر خطأ وقت التشغيل رقم 5 كما تعمدت ترك الكود بلا أي معالجة للخطأ لكي تستطيع مشاهدة ما قمنا بعمله في الدالة Tax .

تمرير الأخطاء في الوحدات النمطية للفئات:

بالرغم من أن كائنات الفئة يمكن أن تقوم بمعالجات للأخطاء التي تحدث في خصائصها وطرقها, إلا أن ذلك يجب ألا يحدث مطلقاً, لأن هذا يربط الكائن وإلى الأبد بتنفيذ إجراء محدد لا فكاك منه, وكما تعلمون أن كائنات الفئة لا تظهر إلى الوجود فجأة, بل يجب أن يتم تمثيلها أولاً عن طريق كود آخر, وهذا الكود هو من يجب عليه أن يستجيب للخطأ الذي يحدث ضمن الفئة, وكمثال قم بتشغيل الكود التالي في وحدة نمطية قياسية:

كود:
Sub testClassError1()
    Dim obj As CSample
    On Error GoTo tstErr
    Set obj = New CSample
    Debug.Print obj.Tax
tstErr_Exit:
    On Error Resume Next
    Set obj = Nothing
tstErr:
    'write here what you want for debugging errors
    Resume tstErr_Exit
End Sub
لاحظ ما يلي:

- لأنه لا توجد معالجة للخطأ في الفئة, فقد قام الفيجوال بيسيك للتطبيقات بتمرير الخطأ من الفئة -حيث حدث الخطأ- إلى الكود الذي قام بالاستدعاء وهو الإجراء testClassError1 الطريقة Err.Raise الموجودة في الفئة.
- ولأننا قمنا بعمل معالجة للخطأ في الإجراء testClassError1 لذلك فلم تظهر أي رسالة خطأ عند التشغيل.

في الموضوع القادم -إن شاء الله- سنتطرق إلى عمر ومجال متغيرات الفئة ومثيلاتها.
عمر و مجال متغيرات الفئة ومثيلاتها
هذا هو الموضوع الثامن عشر في سلسلة البحث المسمى: إنشاء الفئات والبرمجة الموجهة للكائنات
عرفنا في الموضوع السابق كيف يتم تمرير ومعالجة أخطاء الفئات, وأنه يجب على الفئة أن تمرر الخطأ إلى الكود الذي قام بالاستدعاء عن طريق استخدام الطريقة Err.Raise في الفئة, وسواء أكان هذا الخطأ خطأ vba أم كان خطأ من قبل المستخدم, فيجب أن تقوم إجراءات الفئة باصطياده وتمريره إلى الكود المستدعي للفئة والذي يفترض أن يختبر الخطأ ويقوم بمعالجته.

مجال وفترة حياة متغيرات الفئة:

المقصود بمجال المتغير: هو ذلك المكان أو تلك الأمكنة في البرنامج التي يكون فيها المتغير معروفاً و مشاهداً لأكواد ذلك المكان أو تلك الأمكنة. أما المقصود بفترة حياة المتغير: فهو فترة بقاءه و دوامه.
لمتغيرات الفئة نفس مجال و عمر المتغيرات في الوحدات النمطية القياسية, فمثلاً:
المتغيرات على مستوى الفئة:
تكون المتغيرات من النوع Private -والتي يتم الإعلان عنها في قسم التصريحات العامة للفئة- مرئية لجميع الإجراءات الموجودة في تلك الفئة فقط, ويتم تدمير وإنهاء تلك المتغيرات عندما يتم تدمير وإنهاء مثيل الفئة, أما إذا تم الإعلان عن هذه المتغيرات في قسم التصريحات العامة للفئة بأنها من نوع Public فستكون هذه المتغيرات مرئية لأي كود يستطيع الوصول إلى مثيل الفئة.
يعتبر الإعلان عن متغيرات عامة من نوع Public في الفئات طريقة سهلة لإنشاء خصائص الفئات, لكنها طريقة ذات قدرات محدودة خاصة في تطبيق قواعد التحقق من الصحة. للمزيد من المعلومات راجع موضوع إنشاء الخصائص في الفئات.
المتغيرات على مستوى إجراءات الفئة:
أما بالنسبة للمتغيرات التي يتم الإعلان عنها على مستوى إجراءات الفئة, فتبقى هذه المتغيرات مرئية فقط لكود الإجراء الذي هي فيه, وبالتالي فمجالها محدود بذلك الإجراء, لذلك لا تخش حدوث التضارب إذا استخدمت نفس اسم المتغير في عدة إجراءات مختلفة, ففترة حياة المتغير هي نفس فترة حياة الإجراء, ويتم تدميرهذه المتغيرات مع ما تحمله من بيانات عند الخروج من ذلك الإجراء, ويتم تحرير الذاكرة المستخدمة من قبل ذلك المتغير, وإذا تم استدعاء ذلك الإجراء مرة أخرى, فأنه يتم إنشاء المتغير من جديد وكأنه لم يكن له وجود سابق في الذاكرة, ما لم - طبعاً- يتم الإعلان عن هذا المتغير باستخدام الكلمة المفتاحية Static ففي هذه الحالة فأنه يحتفظ بقيمته خلال فترات استدعاءات الإجراء, ويتم تدمير المتغير مع المتغيرات على مستوى الفئة -(وهي تلك التي يتم الإعلان عنها في قسم التصريحات العامة للفئة)- عند تدمير وإنهاء مثيل الفئة.
كود:
Option Compare Database
Option Explicit  'Used to require variable declaration

Public CustomerName As String  'Scope is entire application
Private mVendor As String    'Scope is any procedure in this module
Dim mSupplier As String    'Scope is the current module

Private Sub GetCustomerName()
    Dim strCustomer As String    'Scope is limited to this sub
End Sub
سؤال في الكود السابق أرجو الإجابة عليه .. لماذا اختلف مجال المتغير المسمى mSupplier عن مجال المتغير المسمى strCustomer بالرغم من أن كلاً منهما يبدأ بالكلمة المفتاحية Dim ؟

متغيرات الفئة التي تشير إلى كائنات أخرى:

بالرغم من أن المتغيرات التي تستخدم كمؤشرات للكائنات تخضع لنفس القواعد في مجال وفترة حياة المتغيرات كما تم شرحه آنفاً, إلا أن هذه المتغيرات تتطلب اعتبارات خاصة.

حالات المرجع الدائري Circular Reference Issues :

لقد شاهدنا في موضوع التسلسل الهرمي للفئات كيف يمكن لمثيل في إحدى الفئات أن يشير إلى فئة أخرى, وبالرغم من أن هذه الميزة تعتبر من إحدى نقاط القوة في البرمجة الموجهة للكائنات, إلا أنه يجب الحذر عند تنفيذها خوفاً من الوقوع في شرك الأخطاء التي يصعب اصطيادها أو تشخيصها.
تبدأ المشكلة حينما يبقى مثيلان في حالة مرجع دائري, وذلك عندما تحمل فئة من الفئات مثيلاً يشير إلى فئة أخرى والتي بدورها تحمل مثيلاً يشير إلى الفئة الأولى.
تنص القواعد على أنه لا يمكن تدمير وإنهاء مثيلات الفئة حتى يتم تصفية وتحرير جميع المؤشرات التي تشير إلى تلك المثيلات, ولذلك يحدث أحياناً أن نجد -في التطبيق- مثيلات لا تستجيب لأوامر إنهائها, فمنها ما لا يتم إنهاؤها إلا بشكل متأخر, ومنها ما لا يتم إنهاؤها على الإطلاق, وتبقى عالقة في ذاكرة الجهاز, ويمكن -بالتالي- أن تقود إلى سوء عمل الذاكرة وبطئها. 
وبالمثال يتضح المقال .. سنقوم بإنشاء فئتين ونحفظ الأولى باسم CRef1 والثانية باسم CRef2 والغرض من هاتين الفئتين أن نضع في كل منها مرجعاً يشير إلى الفئة الأخرى ليتوضح معنى المرجع الدائري.
1- كود الفئة الأولى CRef1 :

Option Compare Database
Option Explicit

Private mObjLink As CRef2

Public Property Set Link(objMyObject As CRef2)
    'Create a link from this object to the other one
    Set mObjLink = objMyObject
    Debug.Print "Creating reference to CRef2 from CRef1"
End Property

Private Sub Class_Initialize()
    Debug.Print "Instantiating; CRef1"
End Sub

Private Sub Class_Terminate()
    Debug.Print "Terminating; CRef1; instance"
End Sub
2- كود الفئة الثانية CRef2 :

Option Compare Database
Option Explicit

Private mObjLink As CRef1

Public Property Set Link(objMyObject As CRef1)
    'Create a link from this object to the other one
    Set mObjLink = objMyObject
    Debug.Print "Creating reference to CRef1 from CRef2"
End Property

Private Sub Class_Initialize()
    Debug.Print "Instantiating; CRef2"
End Sub

Private Sub Class_Terminate()
    Debug.Print "Terminating; CRef2; instance"
End Sub
لاحظ ما يلي:
تحتوي كل من الفئتين السابقتين على خاصية Link لإنشاء مثيل فريد للفئة الأخرى, بالإضافة إلى أوامر لطباعة معلومات في النافذة السريعة Immediate window وفيما يلي ستتوضح الحالات.
الإنهاء المتأخر Delayed Termination :
يشمل الاختبار الأول مثيلات فئة يتم إنهاؤها بشكل متأخر, وهذا يحدث عند كتابتك لكود يبقي على مؤشر يشير إلى مثيل لفئة, وهذا المثيل يحتوي -بدوره- على مؤشر يشير إلى فئة أخرى. الإجراء التالي يوضح هذه الحالة:
كود:
Sub TestClassLifeTime1()
    ' Both instances will terminate at the same time
    Dim Aref1 As CRef1
    Dim Aref2 As CRef2
    ' Instantiate variables
    Set Aref1 = New CRef1
    Set Aref2 = New CRef2
    ' Set a reference to one instance from the other
    Set Aref2.Link = Aref1
    ' Destroy the local references
    Set Aref1 = Nothing
    ' Aref1 still exists, it will be terminated with Aref2
    Set Aref2 = Nothing
End Sub
لاحظ ما يلي:

1- يبدأ الإجراء السابق بإنشاء مثيل لكل فئة من الفئات التي أنشأناها سابقاً.
2- يقوم الكود الموجود في حدث Initialize في الفئتين بطباعة معلومات في النافذة السريعة Immediate window .
3- بعد ذلك يقوم الإجراء -عن طريق الخاصية Link- بإعداد مؤشر يشير من المثيل Aref2 إلى المثيل Aref1 . وكذلك تتم طباعة بعض المعلومات الإضافية في النافذة السريعة Immediate window .
4- يحتوي الكود في الإجراء السابق على مؤشرين Aref1 و Aref2 يشير الأول إلى مثيل للفئة CRef1 ويشير الثاني إلى مثيل للفئة CRef2 ويحتوي مثيل الفئة CRef2 على مؤشر يشير إلى مثيل الفئة CRef1 .
5- إن استطعت تنفيذ الإجراء السابق خطوة خطوة -عبر النقر المتكرر على الزر F8- فستجد أنه عند الوصول إلى سطر الكود الذي يقوم بإعداد المثيل Aref1 إلى Nothing فأنه لا يتم كتابة أي شئ في النافذة السريعة Immediate window . إذا جئنا إلى الواقع -للأمانة- فالإجراء ينهي مؤشره الذي يشير إلى CRef1 لكن طالما هناك مؤشر آخر موجود في Aref2 يشير -أيضاً- إلى CRef1 فلا يمكن إنهاء CRef1 . وسيتم إنهاءها فقط عند تنفيذ السطر الأخير من الإجراء في نفس الوقت الذي يتم فيه إنهاء وتدمير CRef2 .
الكائنات اليتيمة Orphaned Objects :
تعتبر هذه الحالة أكثر خطورة من سابقتها لأنها أساس المرجع الدائري, وتمنع إنهاء الكائنات, وهذا سلوك يسبب أضراراً خطيرة لأنه يترك الكائنات في الذاكرة بدون وجود طريقة برمجية لإنهائها, كما أنه يصعب كثيراً معالجتها. يظهر الكود التالي كيفية حدوث هذه الحالة:

Sub TestClassLifeTime2()
    ' No instances will terminate!!
    Dim Aref1 As CRef1
    Dim Aref2 As CRef2
    ' Instantiate variables
    Set Aref1 = New CRef1
    Set Aref2 = New CRef2
    ' Set a reference to one instance from the other
    Set Aref1.Link = Aref2
    Set Aref2.Link = Aref1
    ' Destroy the local references
    Set Aref1 = Nothing
    Set Aref2 = Nothing
    ' Aref1 and Aref2 still exist
End Sub
لاحظ ما يلي:

- الفرق الوحيد بين هذا الإجراء والإجراء الذي سبقه هو سطر واحد فقط وهو الذي يقوم بإنشاء مؤشر يشير من المثيل Aref1 إلى المثيل Aref2 .
- يحتوي الكود في الإجراء السابق على مؤشرين Aref1 و Aref2 يشير الأول إلى مثيل للفئة CRef1 ويشير الثاني إلى مثيل للفئة CRef2 كما يحتوي مثيل كل فئة على مؤشر داخلي يشير إلى مثيل الفئة الأخرى.
- إذا قمت بتنفيذ الكود خطوة خطوة فستلاحظ أنه لا يتم إنهاء أي مثيل من المثيلين بالرغم من إعدادهما بشكل صريح إلى Nothing . فكيف يمكن حدوث ذلك؟ .. يحدث ذلك بسبب بقاء المؤشرات الداخلية في كلا المثيلين بدون أن يتم إنهاؤهما, وسيبقيان في الذاكرة -يشغلانها دون حاجة إليهما- إلى أن يتم إنهاء التطبيق, وأحياناً إلى أن تتم إعادة تشغيل الجهاز.
الإنهاء السليم Proper Termination :
يتطلب تصحيح مشكلة المرجع الدائري أن تولوا انتباهكم إلى احتمال امكانية ظهوره عند إنشاء فئات هرمية. فإذا لم تكن هناك مؤشرات في مثيلات الفئات تشير إلى بعضها بعضاً, فلن يكون هناك مدعاة للقلق من حدوث هذه المشكلة, أما إذا اضطررت إلى عمل ذلك -(يكثر حدوث ذلك في العلاقات بين الفئات خاصة علاقة أب لابن)- فيجب أن تقوم بشكل صريح بإنهاء المؤشرات الداخلية قبل قيامك بإنهاء المؤشرات الخارجية, وفي موضوعنا هذا نقوم بذلك عن طريق إعداد الخاصية Link إلى Nothing قبل إنهاء المثيلات, وهناك مقاربة جميلة لذلك وهي إنشاء طريقة في كل فئة تقوم بإنهاء أي مؤشرات داخلية متبقية, ثم يتم استدعاء هذه الطريقة قبل إنهاء المثيل. وبالمثال يتضح المقال.
1- قم بإضافة الإجراء التالي إلى كلا الفئتين CRef1 و CRef2 .

Public Sub CleanMyLink()
    'If the object exists, destroy it
    If Not mObjLink Is Nothing Then
    Set mObjLink = Nothing
    End If
End Sub
2- قم بإضافة الإجراء التالي إلى وحدة نمطية قياسية وستجد انتهاء مشكلة المرجع الدائري, وبأن الكود يعمل بشكل سليم.

كود:
Sub TestClassLifeTime3()
    ' Objects terminate normally
    Dim Aref1 As CRef1
    Dim Aref2 As CRef2
    ' Instantiate variables
    Set Aref1 = New CRef1
    Set Aref2 = New CRef2
    ' Set a reference to one from the other
    Set Aref1.Link = Aref2
    Set Aref2.Link = Aref1
    ' Clean up internal reference
    Aref1.CleanMyLink
    Aref2.CleanMyLink
    ' Destroy the local references
    Set Aref1 = Nothing
    Set Aref2 = Nothing
End Sub
لعلك لاحظت استدعاء الإجراء CleanMyLink قبل إنهاء المثيلات, وبأن الإجراء TestClassLifeTime3 يعمل بشكل صحيح 100% وقد تم التغلب على مشكلة المرجع الدائري.
في الموضوع القادم -إن شاء الله- سنتطرق إلى الفئات والنماذج فإلى الملتقى.
الفئات والنماذج
هذا هو الموضوع التاسع عشر في سلسلة البحث المسمى: إنشاء الفئات والبرمجة الموجهة للكائنات
رأينا في الموضوع السابق مجال وفترة حياة متغيرات الفئات, وعرفنا أن ذلك يعتمد على الموضع الذي نعلن فيه عن المتغير, وكذلك على الكلمة المفتاحية التي تسبق اسم المتغير, وتطرقنا إلى الإنهاء السليم لمثيلات الفئة ذوات المراجع الداخلية, وأوضحنا أهمية ذلك. واليوم ننتقل إلى شرح استخدام الفئات مع النماذج.

فتح نسخ متعددة من نموذج:

عندما نقوم بإنشاء نموذج لأول مرة, فلا يكون للنموذج الجديد وحدة نمطية مترافقة معه, ولكي نستطيع أن نفتح نسخاً متعددة من نموذج, فيجب أن يكون لذلك النموذج وحدة نمطية -حتى ولو كانت فارغة- وإلا فلا يمكننا فتح نسخ متعددة من نفس النموذج. 

كيفية إنشاء وحدة نمطية لنموذج:

لا يوجد أسهل من هذا الأمر. تحتاج فقط إما إلى إظهار نافذة الكود خلف النموذج المعني وبمجرد ظهور تلك النافذة فسيصبح لذلك النموذج وحدة نمطية خاصة به, وإما إلى تغيير قيمة خاصية النموذج المدعوة: "لديه وحدة نمطية" من لا إلى نعم وبذلك يحصل المطلوب.

طريقة فتح نسخ متعددة من نموذج:

إذا أردنا فتح نسخة واحدة فقط من نفس النموذج المفتوح, فسنحتاج إلى متغير واحد من نوع نموذج, أما إذا أردنا فتح نسخ متعددة من نفس النموذج, فسنحتاج إما إلى عدة متغيرات من نوع نموذج, أو -بدلاً عن ذلك- مجموعة يمكنها أن تحمل المؤشرات إلى النموذج كما يلي:
1- لنفرض أن لدينا نموذجاً باسم frm1 وأردنا أن نفتح عدة نسخ منه في نفس الوقت.
2- قم بإنشاء نموذج ثان وضع فيه زر أمر باسم cdOpenFrm واحفظ النموذج باسم frmMain .
3- في قسم التصريحات العامة للنموذج frmMain قم بالتصريح عن كائن مجموعة باسم colForms كما يلي:
ود:
Dim colForms As New Collection
4- ضع الكود التالي في حدث عند النقر للزر cdOpenFrm :
و:
Private Sub cdOpenFrm_Click()
    Dim frm As Form_frm1
    Set frm = New Form_frm1
    colForms.Add frm, CStr(CLng(Rnd * (2 ^ 31)))
    frm.Visible = True
End Sub
لاحظ ما يلي:

- تم الإعلان في السطر الأول من الكود السابق عن متغير من نوع النموذج المراد فتح عدة نسخ منه, ولو لم يكن للنموذج المسمى frm1 وحدة نمطية لكان السطر الأول من الكود مرفوضاً من قبل الفيجوال بيسيك للتطبيقات جملة وتفصيلاً.
- في السطر الثاني تم إعداد المتغير ليكون مثيلاً جديداً للنموذج المذكور.
- في السطر الثالث أضفنا مثيل النموذج إلى كائن المجموعة مع توليد رقم عشوائي كمفتاح غير قابل للتكرار تقريباً, لماذا تقريباً؟!! .. لأن الرقم العشوائي الذي سيتم توليده سيكون رقماً صحيحاً طويلاً يقع بين الرقم 215 و الرقم 2,147,483,433 ملياران ومئة وسبعة وأربعون مليوناً أو يزيد, ولذلك فاحتمال تكرار نفس الرقم مرتين شبه مستحيل.
- في السطر الرابع نجد الأمر الصريح بإظهار النسخة المثيلة من النموذج, وكلما نقرت على الزر cdOpenFrm كلما ظهرت أمامك نسخة جديدة من النموذج, ولن يمل الزر حتى تمل أنت.

5- وأخيراً ضع الكود التالي في حدث عند الإغلاق أو عند إلغاء التحميل للنموذج frmMain :

كود:
Dim frm As Form_frm1
For Each frm In colForms
    Set frm = Nothing
Next
ولا يخفى على شريف علمكم معنى هذا الكود ولا فائدته.

كيفية الإشارة بالكود إلى نسخ متعددة مفتوحة من نموذج:

يجب الإشارة إلى أنه في حالة فتح عدة نسخ من نفس النموذج فأن بعض الصيغ في الأكواد لن تعمل بشكل صحيح, ومثال ذلك الصيغة التالية لن تعمل بشكل صحيح:

كود:
Forms!frm1.Requery
و سنحتاج من أجل أن يعمل السطر السابق بشكل صحيح إلى استعمال فهرس عددي للمجموعة كالتالي:

كود:
Forms(2).Requery
أو أن نستخدم متغيراً -من نوع نموذج- خاصاً بنا.

ملاحظة: 
سيتم إضافة عدة مشاركات إلى هذا الموضوع لاستكماله, وذلك لأهمية هذا الموضوع, ولأن هذا الموضوع قد يطول شرحه, ولأنني لا أملك الكثير من الوقت ولا الجهد كي أسرده كاملا في مشاركة واحدة, فأرجو منكم أن تصبروا, وتصفحوا, وتعفوا عن محبكم, والكريم من عفا.
السلام عليكم ورحمة الله وبركاته
أشكركم أساتذتي وإخوتي على شكركم لي
نتابع ما سبق
يمكن لنا أن نفتح نسخاً متعددة من تقرير بنفس الطريقة المشروحة آنفاً في فتح نسخ متعددة من نموذج, ولا داعي لتكرار الشرح.
الفئات والنماذج المرتبطة:
يحلو لبعض المبرمجين أن يقوموا بربط مجموعة السجلات Recordset التي تظهر على النموذج باستخدام الكود البرمجي, وأسهل طريقة لعمل ذلك هي أن تقوم بإنشاء مجموعة سجلات Recordset ثم تأتي إلى حدث فتح النموذج وتقوم بإسناد قيمة خاصية النموذج المسماة Recordset إلى مجموعة السجلات التي قمت بإنشائها, وبالمثال يتضح المقال.
ربط نموذج إلى مجموعة سجلات بتقنية الـ DAO :
1- نقوم أولاً بإنشاء فئة ونحفظها باسم مثلاً COrdersDAO وهي الفئة التي ستربط النموذج إلى جدول معين في قاعدة البيانات, وليكن اسم الجدول مثلاً tblOrders . لاحظ أن: جميع الأسماء الواردة في هذا الموضوع للمتغيرات والفئات والنماذج والكائنات سابقاً ولاحقاً هي أسماء اختيارية يمكنك تغييرها بما تشاء.
2- ضع الكود التالي في الفئة التي قمت بحفظها:

كو:
Option Compare Database
Option Explicit

Private mDb As Database
Private mRec As Recordset

Private Sub Class_Initialize()
    Set mDb = CurrentDb
    Set mRec = mDb.OpenRecordset("tblOrders", dbOpenDynaset)
End Sub

Public Property Get Recordset() As Recordset
    Set Recordset = mRec
End Property

Private Sub Class_Terminate()
    mRec.Close: Set mRec = Nothing
    mDb.Close: Set mDb = Nothing
End Sub
لاحظ ما يلي:

- تم الإعلان في قسم التصريحات العامة للفئة عن متغيرين خاصين بالفئة, الأول من نوع كائن قاعدة بيانات والثاني من نوع كائن مجموعة سجلات, والمتغيران مرئيان لأحداث وإجراءات الفئة فقط.
- في حدث Initialize -والذي سيتم قدحه عند إنشاء مثيل جديد للفئة- تم إعداد المتغير الأول ليشير إلى قاعدة البيانات الحالية, وإعداد المتغير الثاني ليشير إلى مجموعة سجلات جدول الطلبيات.
- تم إنشاء خاصية جديدة باسم Recordset لتكون واجهة للفئة, وتم إعدادها لتشير إلى المتغير الثاني.
وأهمية هذه الخاصية أنها تجعل التعامل مع الفئة ممكناً, بحيث تستطيع الأكواد خارج الفئة أن ترى وتسمع وتعلم ما يدور داخل مثيل وشبيه الفئة, وبالتالي تحدث الاستجابة والتفاعل بين أكواد الإجراء الذي قام باستدعاء الفئة من جهة, وبين أكواد الفئة نفسها من جهة أخرى.
- في حدث Terminate -والذي سيتم قدحه عند إنهاء مثيل الفئة- سيتم حينها إغلاق وإنهاء مجموعة السجلات وقاعدة البيانات ومحوهما من الذاكرة.
3- الآن قم بوضع الكود التالي في نموذج الطلبيات:
كود:
Option Compare Database
Option Explicit

Private Orders As COrdersDAO

Private Sub Form_Open(Cancel As Integer)
    Set Orders = New COrdersDAO
    Set Me.Recordset = Orders.Recordset
End Sub

Private Sub Form_Close()
    Set Orders = Nothing
End Sub
لاحظ ما يلي:

- في قسم التصريحات العامة للنموذج تم الإعلان عن متغير خاص من نوع الفئة التي أنشأناها سابقاً.
- في حدث عند الفتح للنموذج تم إعداد ذلك المتغير ليشير إلى مثيل جديد للفئة, وكذلك تم إعداد وإسناد قيمة خاصية النموذج المسماة Recordset إلى الخاصية Recordset الموجودة في مثيل الفئة.
- في حدث عند إغلاق النموذج يتم إنهاء مثيل الفئة بهدف تنظيف المخلفات من الذاكرة أولاً بأول.
- من أجل أن يعمل الكود السابق بشكل صحيح, يجب أن تكون أسماء ومصدر بيانات مربعات النص في نموذج الطلبيات متطابقة مع أسماء الحقول في جدول الطلبيات, إلا أن مصدر بيانات النموذج نفسه يكون خالياً.
قم بتطبيق ما سبق في مثال من عندك وإن واجهت أي خطأ فلنناقشه معاً.
يتبع.........
نتابع ما سبق
في المثال السابق قمنا - بشكل مبسط جداً - بربط نموذج إلى مجموعة سجلات كنا قد أنشأناها مسبقاً في الفئة المسماة COrdersDAO لتكون الفكرة قريبة من الأذهان. 
قد تكون أكواد الفئة - أحياناً- أكثر تعقيداً, والمبدأ هو نفسه كالسابق ولكنه أكثر تفصيلاً, حيث تحتوي الفئة -بالإضافة إلى ما سبق- على واجهات تقوم أحدها بالإضافة, وأخرى بالتحديث, وثالثة بالحذف, ورابعة بالبحث, وما إلى ذلك مما يطلبه المستخدمون, وسنحتاج -في مثال يشرح ذلك- إلى إنشاء جدول وحفظه باسم tblOrders حيث يحتوي هذا الجدول على الحقول التالية:
- الحقل OrderID نوعه ترقيم تلقائي.
- الحقل CustomerID نوعه عدد صحيح طويل
- الحقل Item نوعه نص.
- الحقل Qty نوعه عدد صحيح طويل.
- الحقل UnitPrice نوعه عملة.
قم بإدخال بيانات في الجدول لبضعة سجلات, ثم قم بإنشاء وحدة نمطية للفئة واحفظها باسم COrdersDAO2 وضع فيها الكود التالي:
Option Compare Database
Option Explicit

Private mOrderID As Long
Public CustomerID As Long
Public Item As String
Public Qty As Long
Public UnitPrice As Currency

Property Get OrderID() As Long      'Read only Property
    OrderID = mOrderID
End Property

Public Function FindOrder(ID As Long) As Boolean
    On Error GoTo HandleError
    
    Dim rs As DAO.Recordset
    Dim strSQL As String
    
    FindOrder = False
    
    strSQL = "SELECT * FROM tblOrders WHERE ([OrderID]=" & ID & ");"
    
    Set rs = CurrentDb.OpenRecordset(strSQL, dbOpenForwardOnly)
    With rs
        If .RecordCount = 0 Then
            MsgBox "Cannot find Order with ID = " & ID, vbCritical
            GoTo Exit_Function
        End If
        
        ' Use Nz function in case of Null values in the table
        mOrderID = Nz(!OrderID)
        Me.CustomerID = Nz(!CustomerID)
        Me.Item = Nz(!Item)
        Me.Qty = Nz(!Qty)
        Me.UnitPrice = Nz(!UnitPrice)
        
        .Close
    End With
    FindOrder = True
    
Exit_Function:
    Exit Function
HandleError:
    MsgBox "Error " & Err.Number & " can't find Order No:" & vbCrLf & _
    Err.Description, vbCritical
    Resume Exit_Function

End Function

Public Function NewOrder() As Boolean
    On Error GoTo HandleError
    
    NewOrder = False
    
    Dim rs As DAO.Recordset
    
    Set rs = CurrentDb.OpenRecordset("tblOrders", dbOpenDynaset)
    With rs
        .AddNew
        SaveRecord rs
        .Update
        
        .Bookmark = .LastModified
        mOrderID = !OrderID   ' Get ID from new record
        .Close
    End With
    Set rs = Nothing
    
    NewOrder = True
    
Exit_Function:
    Exit Function
    
HandleError:
    MsgBox "Error " & Err.Number & " in NewOrder" & vbCr & Err.Description, vbCritical
    Resume Exit_Function
    
End Function

Private Sub SaveRecord(rs As DAO.Recordset)
    With rs
        !CustomerID = Me.CustomerID
        !Item = Me.Item
        !Qty = Me.Qty
        !UnitPrice = Me.UnitPrice
    End With
End Sub

Public Function UpdateRecord() As Boolean
    On Error GoTo HandleError
    
    UpdateRecord = False
    
    Dim rs As DAO.Recordset
    Dim strSQL As String
    
    strSQL = "SELECT * FROM tblOrders WHERE ([OrderID]=" & Me.OrderID & ");"
    Set rs = CurrentDb().OpenRecordset(strSQL, dbOpenDynaset)
    With rs
        If .RecordCount = 0 Then
            .Close
            Err.Raise 40000, , "Cannot find OrderID=" & Me.OrderID
            GoTo Exit_Function
        End If
        
        .Edit
        SaveRecord rs
        .Update
        
        .Close
    End With
    Set rs = Nothing
    
    UpdateRecord = True
    
Exit_Function:
    Exit Function
    
HandleError:
    MsgBox "Error " & Err.Number & " in Update" & vbCr & Err.Description, vbCritical
    Resume Exit_Function
End Function

Public Function DeleteOrder() As Boolean
    On Error GoTo HandleError
    Dim strSQL As String
    DeleteOrder = False
    
    If Me.OrderID <= 0 Then
        GoTo Exit_Function
    End If
    
    strSQL = "Delete * FROM tblOrders WHERE ([OrderID]=" & Me.OrderID & ");"
    CurrentDb.Execute strSQL, dbFailOnError
    
    DeleteOrder = True
Exit_Function:
    Exit Function
HandleError:
    MsgBox "Error " & Err.Number & " in DeleteOrder" & vbCr & Err.Description, vbCritical
    Resume Exit_Function
End Function
لاحظ ما يلي:
- تحتوي الفئة في قسم التصريحات العامة على متغيرات خاصة ومتغيرات عامة, حيث تعمل المتغيرات العامة -بشكل مباشر- كخصائص وواجهات للفئة.
- كذلك هناك إجراءات خاصية, وإجراءات دوال تعمل أيضاً كواجهات للفئة, ولاحظ الفرق بين إجراء الخاصية وإجراء الدوال في عمل كل منهما وفي حجم ومقدار الأكواد فيهما.
أخيراً ومن أجل اختبار الفئة سنحتاج إلى إنشاء نموذج يحتوي على ما يلي:
مربعات نص غير منضمة واجعل أسماءها كالتالي:
- txtFind
- txtCustomer
- txtItem
- txtQty
- txtUnitPrice
كذلك سنضيف إلى النموذج زريّ أمر بحيث يكون اسم الزر الأول cdFind و اسم الزر الثاني cmdAdd . (لا داعي لتذكيركم بأنه يمكنكم تغيير جميع الأسماء الواردة كما تحبون).
في قسم التصريحات العامة للنموذج نضع الكود التالي::
Private mOrder As New COrdersDAO2
ونضع الكود التالي في حدث عند النقر للزر الأول cdFind :
Private Sub cdFind_Click()
    If Not IsNumeric(txtFind) Then
        MsgBox "Sorry Order Id is not valid."
        Exit Sub
    Else
        mOrder.FindOrder (txtFind)
        Me.txtCustomer = mOrder.CustomerID
        Me.txtItem = mOrder.Item
        Me.txtQty = mOrder.Qty
        Me.txtUnitPrice = mOrder.UnitPrice
    End If
End Sub
ونضع الكود التالي في حدث عند النقر للزر الثاني cmdAdd :
Private Sub cmdAdd_Click()
    On Error GoTo HandleError
    If IsNull(Me.txtCustomer) Then
        MsgBox "You must enter a customer"
        Exit Sub
    Else
        mOrder.CustomerID = Me.txtCustomer
    End If
    
    If IsNull(Me.txtItem) Then
        MsgBox "You must enter a an item"
        Exit Sub
    Else
        mOrder.Item = Me.txtItem
    End If
    
    If IsNull(Me.txtQty) Then
        MsgBox "You must enter  a quantity"
        Exit Sub
    Else
        mOrder.Qty = Me.txtQty
    End If
    
    If IsNull(Me.txtUnitPrice) Then
        MsgBox "You must enter a price"
        Exit Sub
    Else
        mOrder.UnitPrice = Me.txtUnitPrice
    End If
    
    ' All gathered, so let's prompt the user
    If vbOK <> MsgBox("Do you want to save this order?", vbOKCancel) Then
        Exit Sub
    End If
    
    ' Got here, so let's save it
    If Not mOrder.NewOrder Then
        MsgBox "Unable to create a new order", vbCritical
        Exit Sub
    Else
        MsgBox "Successfully Created new order with ID=" & mOrder.OrderID
    End If
    
ExitAdd:
    Exit Sub
HandleError:
    MsgBox "Error: " & Err.Number & vbCr & Err.Description
    Resume ExitAdd
    
End Sub
ولا ننسى أن نقوم بتنظيف الذاكرة بوضع الكود التالي عند إغلاق النموذج:
Private Sub Form_Close()
    Set mOrder = Nothing
End Sub
تذكر ما يلي:
- أن مصدر بيانات النموذج فارغ.
- أن جميع مربعات النص في النموذج غير منضمة.
- أنه يمكنك بنفس الطريقة السابقة عمل زر للتحديث وآخر للحذف.
قم الآن بتجربة النموذج بالإضافة تارة, وبالبحث تارة أخرى, وأطلعني على ما يستجد لديك.
قم بإضافة زر إلى النموذج يقوم بحذف سجل من الجدول, وأطلعني على الكود.
يتبع .....
مسك الختام:
نتابع ما سبق ونختم هذا البحث بهذا الموضوع المهم والمكثف والذي سنحاول من خلاله أن نتطرق إلى نقاط مهمة تغافلنا عنها سابقاً وتندرج ضمن الأساسيات التي يجب معرفتها وقد عرفنا معظمها سابقاً, ومن هذه النقاط مثلاً كيف تشير الفئة إلى نفسها؟ .. وماهي الدالة ObjPtr وكيف نستفيد منها في الفئات؟!, كما سنطبق معظم ما تعلمناه من بداية هذا البحث إلى ختامه, حيث سنتعلم كيف نستطيع:
1- استخدام مجموعة من الفئات collection of classes .
2- إعداد فئة المجموعة class collection كخاصية لفئة أخرى, مثل الخاصية المسماة Invoices والموجودة في الفئة المسماة CCustomer وستجدها بالشكل التالي:
كود:
Public Property Set Invoices(ByVal clsInvoices As CcolInvoices): Set mCInvoices = clsInvoices: End Property
3- إعداد خاصية Parent عبر تحويل متغير نوع بياناته "عدد صحيح طويل Long " إلى متغير نوع بياناته "كائن Object " وذلك الدالة المسماة ObjPtr , ومن ثم الحصول على قيمة الخاصية Parent عبر دالة معرفة من قبل المستخدم أسميناها ObjectFromPattern ولن تخطئها -حين تراها- عينك.
4- استخدام فئة المجموعة المتداخلة nested class collection بما يشبه الجدول الأساسي (الأب) والجدول الفرعي (الابن) وفي هذا المثال فإن الفئة المسماة CInvoice تمثل الجدول الابن, والفئة المسماة CCustomer تمثل الجدول الأب.
كيف تشير الفئة إلى نفسها؟:
تعلمون أنه يمكن للفئة أن تشير إلى نفسها عبر الكلمة المفتاحية Me تماماً كما في الوحدات النمطية للفئة الخاصة بالنماذج, لكن ما لا تعلمونه -ربما- هو أنه يمكن لنا أن نصرح في الفئة عن متغير يكون نوع بياناته من نفس نوع الفئة التي هو فيها, بحيث أنه عند إعداد مثيل جديد لهذا المتغير, فأننا نستطيع القول بأن الفئة تشير إلى نفسها, وهذا هو ما يعطي الاكسس القدرة على التعامل مع هياكل البيانات الديناميكية أمثال الطوابير والمكدسات, حيث يلعب هذا المتغير الذي يشير إلى نفس الفئة دوراً حيوياً كمرجع يشير إلى العنصر الأخير في قمة المكدس. لن نتطرق هنا إلى موضوع الفئات وهياكل البيانات, لأنه يحتاج -على الأقل- إلى بحث مستقل إن لم يكن بحثين مستقلين, يبحث الأول في موضوع هياكل البيانات, ويبحث الثاني في موضوع الفئات وهياكل البيانات.
فحص كائن أو تركيب Structure بالدالة TypeOf :
التركيب عبارة عن نمط معرف من قبل المستخدم, ويمكنك اعتباره شكلاً مبسطاً من الفئات حيث يحمل التركيب متغيرات كالفئات إلا أنه لا يحمل إجراءات مثلها. نستطيع اختبار نوع بيانات متغير ما عن طريق دالة مبنية في الفيجوال بيسيك TypeOf ... Is وتستطيع هذه الدالة تمييز نوعين من البيانات فقط وهما: البيانات من نوع تركيب, والبيانات من نوع كائن, بينما لا تفيد هذه الدالة في فحص المتغيرات التي تحمل أنواع بيانات أخرى, وكمثال على هذه الدالة, ضع هذا الكود في وحدة نمطية قياسية ثم قم بتشغيله:
كود:
Option Compare Database
Option Explicit

Type APoint
    x As Double
    y As Double
End Type

Sub Test_Type()
    Dim Point1 As APoint
    Dim Point2 As APoint
    Point1.x = 4
    Point1.y = -5
    Point2 = Point1 ' 'Make a copy
    Point2.x = 3
    Point2.y = -6
    Debug.Print Point1.x, Point1.y ' 4, -5: not overriden with 3 and -6
    Debug.Print TypeOf Point1 Is APoint ' 'True
    Debug.Print TypeOf Point1 Is Object  ' 'False
End Sub
لاحظ ما يلي:
- يجب أن يتم تعريف النمط خارج الإجراء قبل أن يتم استخدامه في الإجراء.
- المتغير الذي يحمل نوع بيانات من نمط التركيب لا يعتبر كائناً, ولذلك لم يطرأ تغيير على قيمة x أو قيمة y في المتغير Point1 عند تغيير قيمتهما في المتغير Point2 .
- لا تنس دوال الاختبار الأخرى مثل IsObject و TypeName .
كيف تنشئ العلاقات بين الفئات؟:
1- قم بإنشاء فئتين, واحدة لجدول الأب CCustomer وأخرى لجدول الابن CInvoice .
2- ضع في كل فئة خصائصها وطرقها -كما تعلمت سابقاً- مع إضافة خاصية Parent -كما سترى- في كلا الفئتين.
3- قم بإنشاء فئتي مجموعة collection classes الفئة الأولى لكي تجمع كائنات الأب, وإن شئت فقل -صادقاً غير مكذوب- سجلات جدول الأب, وتجد هذه الفئة هنا تحت اسم CcolCustomers , والفئة الثانية لكي تجمع كائنات الابن, أو بالأحرى سجلات الجدول الفرعي (الابن), وستجد فئة المجموعة هذه أدناه تحت اسم CcolInvoices .
4- أضف إلى قسم التصريحات العامة في فئة الأب متغيراً من نوع فئة مجموعة الابن.
5- استخدم الحدثين Initialize و Terminate الموجودين في فئة الأب لإعداد وإنهاء المتغير الذي أضفته في الخطوة رقم 4 , حيث سيتم إنشاء مجموعة تضم الأبناء في كل مرة يتم فيها إنشاء مثيل جديد لفئة الأب, كما سيتم تدمير وإنهاء مجموعة الأبناء عند إنهاء وتدمير مثيل فئة الأب.
6- يتم إعداد وإسناد قيمة الخاصية Parent لفئة الابن في فئة المجموعة التي تضم الأبناء وذلك في الدالة المسماة Add في السطر الذي يسبق -مباشرة- الأمر بضم كائن الفئة الابن إلى المجموعة كما يلي:

Set clsInvoice.Parent = Me
mcolInvoices.Add clsInvoice, CStr(clsInvoice.InvoiceId)
حيث أن clsInvoice وسيطة في الدالة Add كالتالي:
Public Sub Add(clsInvoice As CInvoice)
كود الفئة CCustomer :
Option Compare Database
Option Explicit

'Code of CCustomer class
Private mCustomerId As Long
Private mCustomerName As String
Private mAddress As String
Private mCInvoices As CcolInvoices
Private mParent As Long
Private Declare Sub CopyMemory Lib "kernel32" Alias "RtlMoveMemory" _
    (dest As Any, Source As Any, ByVal bytes As Long)


Public Property Set Invoices(ByVal clsInvoices As CcolInvoices): Set mCInvoices = clsInvoices: End Property
Public Property Get Invoices() As CcolInvoices: Set Invoices = mCInvoices: End Property
Public Property Let CustomerId(ByVal lngCustomerId As Long): mCustomerId = lngCustomerId: End Property
Public Property Get CustomerId() As Long: CustomerId = mCustomerId: End Property
Public Property Let CustomerName(ByVal strCustomerName As String): mCustomerName = strCustomerName: End Property
Public Property Get CustomerName() As String: CustomerName = mCustomerName: End Property
Public Property Let Address(ByVal strAddress As String): mAddress = strAddress: End Property
Public Property Get Address() As String: Address = mAddress: End Property
Public Property Get Parent() As CcolCustomers: Set Parent = ObjectFromPattern(mParent): End Property
Public Property Set Parent(obj As CcolCustomers): mParent = ObjPtr(obj): End Property
'ObjPtr return a long value for object except custom objects, it returns object name
Private Function ObjectFromPattern(ByVal pObj As Long) As Object
    Dim obj As Object
    CopyMemory obj, pObj, 4
    Set ObjectFromPattern = obj
    ' manually destroy the temporary object variable
    CopyMemory obj, 0&, 4
End Function

Private Sub Class_Initialize()
    Set mCInvoices = New CcolInvoices
End Sub

Private Sub Class_Terminate()
    Set mCInvoices = Nothing
End Sub
لاحظ ما يلي:

- في قسم التصريحات العامة للفئة صرحنا عن دالة من دوال الـ api وهي الدالة CopyMemory والتي نحتاج إليها في إعداد خاصية الفئة المسماة Parent
- تم الإعلان أيضاً عن متغير mCInvoices من نوع فئة المجموعة التي تضم الأبناء CcolInvoices , حيث استخدمنا فئة المجموعة في إعداد خاصية الفئة CCustomer المسماة Invoices .
- تم الإعلان عن متغير mParent من نوع Long ليحمل معرفاً للخاصية Parent , وقد استخدمنا الدالة المبنية ObjPtr في الحصول على قيمة هذا المعرف.
ما هي الدالة ObjPtr وما أصلها؟:
الدالة ObjPtr هي دالة مخفية في الفيجوال بيسيك للتطبيقات, وتعيد هذه الدالة العنوان (المعرف) الموجود في الذاكرة لمتغير من نوع كائن, كما يمكن استخدامها في اختبار متغيرين من نوع كائن فيما إذا كان هذان المتغيران يشيران إلى كائن واحد, أم لا؟.
- أضفنا إلى الفئة دالة أسميناها ObjectFromPattern تقوم بعكس الدالة ObjPtr إذ نمرر إليها متغيراً من نوع Long وهو العنوان المعرف للكائن, فتعيد إلينا ذلك الكائن بشحمه ولحمه.
كود الفئة CInvoice :
كود:
Option Compare Database
Option Explicit

'Code of CInvoice class
Private mInvoiceId As Long
Private mQty As Double
Private mPrice As Double
Private mItem As String
Private mInvoiceDate As Date
Private mlYear As Long
Private mlMonth As Long
Private mParent As Long
Private Declare Sub CopyMemory Lib "kernel32" Alias "RtlMoveMemory" _
    (dest As Any, Source As Any, ByVal bytes As Long)

Public Property Let InvoiceId(ByVal lngInvoiceId As Long): mInvoiceId = lngInvoiceId: End Property
Public Property Get InvoiceId() As Long: InvoiceId = mInvoiceId: End Property
Public Property Let Qty(ByVal dQty As Double): mQty = dQty: End Property
Public Property Get Qty() As Double: Qty = mQty: End Property
Public Property Let InvoiceDate(ByVal dtInv As Date): mInvoiceDate = dtInv: End Property
Public Property Get InvoiceDate() As Date: InvoiceDate = mInvoiceDate: End Property
Public Property Let Price(ByVal dPrice As Double): mPrice = dPrice: End Property
Public Property Get Price() As Double: Price = mPrice: End Property
Public Property Let Item(ByVal strItem As String): mItem = strItem: End Property
Public Property Get Item() As String: Item = mItem: End Property
Public Property Set Parent(obj As CcolInvoices): mParent = ObjPtr(obj): End Property
Public Property Get Parent() As CcolInvoices: Set Parent = ObjectFromPattern(mParent): End Property

Private Function ObjectFromPattern(ByVal pObj As Long) As Object
    Dim obj As Object
    CopyMemory obj, pObj, 4
    Set ObjectFromPattern = obj
    ' manually destroy the temporary object variable
    CopyMemory obj, 0&, 4
End Function

Function ItemTotal() As Double
    ItemTotal = Me.Qty * Me.Price
End Function
لاحظ ما يلي:
- أن خاصية Parent موجودة في كلا الفئتين, مع ما يتبعها من دوال كما شرحناها سابقاً.
- عند تمثيل الفئة CInvoice فأن كل مثيل من الفئة يقابل -في عالم قواعد البيانات- سجلاً واحداً من سجلات تفاصيل الفاتورة, أما جميع أصناف الفاتورة الواحدة فسيضمها مثيل فئة المجموعة المسماة CcolInvoices وهذا هو الكود الخاص بها:
كود:
Option Compare Database
Option Explicit

'Code of CcolInvoices class
Private mcolInvoices As Collection

Private Sub Class_Initialize()
    Set mcolInvoices = New Collection
End Sub

Private Sub Class_Terminate()
    Set mcolInvoices = Nothing
End Sub

Public Function NewEnum() As IUnknown
    Set NewEnum = mcolInvoices.[_NewEnum]
End Function

'first way to add member to a collection
'look how the class refers to itself by Me keyword.
'look also how to use the child's class parent property in this method.
Public Sub Add(clsInvoice As CInvoice)
    If clsInvoice.InvoiceId = 0 Then
        'creating a default InvoiceId if there is no one
        clsInvoice.InvoiceId = Me.Count + 1
    End If

    Set clsInvoice.Parent = Me
    mcolInvoices.Add clsInvoice, CStr(clsInvoice.InvoiceId)
End Sub

'to get an item from the collection
Public Property Get Invoice(vItem As Variant) As CInvoice
    Set Invoice = mcolInvoices.Item(vItem)
End Property

Public Property Get Count() As Long
    Count = mcolInvoices.Count
End Property

'another way to add direct data to the collection as a class member
'there is no need to use Parent property here because this method has Add method at the last line.
Public Sub AddInvoice(strItem As String, ByVal dQty As Double, ByVal dPrice As Double, ByVal dtInv As Date)

    Dim clsInvoice As CInvoice

    Set clsInvoice = New CInvoice
    With clsInvoice
        .Item = strItem
        .Qty = dQty
        .Price = dPrice
        .InvoiceDate = dtInv
    End With

    Me.Add clsInvoice

End Sub
لاحظ ما يلي:
- كيف تشير فئة المجموعة إلى نفسها عبر الكلمة المفتاحية Me . وكيفية استخدام خاصية Parent الموجودة في الفئة CInvoice لتربط الفئتين معاً بحيث يرتبط مثيل واحد من فئة المجموعة مع مثيل واحد أو مثيلات متعددة من الفئة CInvoice وكأننا -تماماً- أمام علاقة واحد إلى كثير.
- هناك طريقتين لإضافة كائنات الفئة CInvoice إلى فئة المجموعة, الطريقة الأولى عبر الإجراء Add , وهي الطريقة الأساسية للإضافة, والطريقة الثانية عبر الإجراء AddInvoice وهي طريقة اختيارية يمكنك إضافتها -إن شئت- لتسهيل الإدخالات فقط لا غير.
- لقد قمت بإعداد فئتي المجموعة عبر التصدير وإعادة الاستيراد كما شرحناها في المواضيع السابقة لكي أتمكن من التجربة العملية للأكواد.

والفئة الأخيرة هي فئة المجموعة CcolCustomers وهذا كودها:
كود:
Option Compare Database
Option Explicit

'Code of CcolCustomers class
Private mcolCustomers As Collection

Private Sub Class_Initialize()
    Set mcolCustomers = New Collection
End Sub

Private Sub Class_Terminate()
    Set mcolCustomers = Nothing
End Sub

Public Function NewEnum() As IUnknown
    Set NewEnum = mcolCustomers.[_NewEnum]
End Function

Public Sub Add(clsCustomer As CCustomer)
    If clsCustomer.CustomerId = 0 Then
        clsCustomer.CustomerId = Me.Count + 1
    End If

    Set clsCustomer.Parent = Me
    mcolCustomers.Add clsCustomer, CStr(clsCustomer.CustomerId)
End Sub

'to get an item from the collection
Public Property Get Customer(vItem As Variant) As CCustomer
    Set Customer = mcolCustomers.Item(vItem)
End Property

Public Property Get Count() As Long
    Count = mcolCustomers.Count
End Property
الآن لنقم بتجربة جميع الفئات السابقة في وحدة نمطية قياسية, ولنضع فيها الكود التالي ثم لنقم بتشغيله:

كود:
Sub Test_Class()
    Dim clsCustomers As CcolCustomers
    Dim clsCustomer As CCustomer
    Dim clsInvoice As CInvoice
'there is no object variable for Ccolinvoices collection class, because Ccolinvoices exists inside CCustomer class

    Set clsCustomers = New CcolCustomers
    Set clsCustomer = New CCustomer
    clsCustomer.CustomerName = "Aref Hassan"
    clsCustomer.Address = "Yemen - Tiaz"

    'main way to add an Invoice
    Set clsInvoice = New CInvoice
    clsInvoice.Item = "Books"
    clsInvoice.Qty = 1000
    clsInvoice.Price = 3.5
    clsInvoice.InvoiceDate = #9/18/2014#
    
'Invoices is a property in CCustomer class defined as Ccolinvoices class collection, it has Add & AddInvoice methods
    clsCustomer.Invoices.Add clsInvoice

    'Quickway to add an Invoice
    clsCustomer.Invoices.AddInvoice "Pens", 2000, 2.8, #9/18/2014#
'you can use it like this:
    'clsCustomer.Invoices.AddInvoice rs!Item, rs!Qty, rs!Price, rs!Date
    
'Now add the father's record to the father's collection class
    clsCustomers.Add clsCustomer

    For Each clsCustomer In clsCustomers
        For Each clsInvoice In clsCustomer.Invoices
            Debug.Print clsCustomer.CustomerName, clsCustomer.Address, clsInvoice.Item, _
            clsInvoice.Qty, clsInvoice.InvoiceDate, clsInvoice.ItemTotal
        Next clsInvoice
        'if you put next line outside the loop, you will get a runtime error (object not defined)
        'because each clsInvoice instance is terminated when clsCustomer is finished
        Debug.Print "Invoices items = " & clsCustomer.Invoices.Count
    Next clsCustomer
    
    Debug.Print "clsCustomers count = " & clsCustomers.Count
    
    Set clsCustomers = Nothing
    Set clsCustomer = Nothing
    Set clsInvoice = Nothing
End Sub
لاحظ ما يلي:
- لم يتم الإعلان عن متغير من نوع فئة مجموعة الأبناء CcolInvoices وذلك لأنها موجودة كخاصية ضمن الفئة CCustomer باسم Invoices حيث يتم هناك إعداد وتجهيز فئة مجموعة الأبناء.
- هناك طريقتان لإضافة مثيل فئة الابن CInvoice إلى فئة مجموعة الأبناء CcolInvoices كما هو واضح في الكود أعلاه.
- لاحظ طريقة استخدام حلقة الدوران For Each كرّتين متداخلتين, الخارجية منهما لبيانات الجدول الأب, والداخلية لبيانات الجدول الابن.

أتمنى أن أكون قد استطعت -بهذا الشرح المتواضع- تقريب معنى الفئات والبرمجة الموجهة للكائنات إلى أذهانكم, وأن يكون مفهوم الفئات قد أصبح أليفاً إلى قلوبكم, متمنياً أن لا تكتفوا بالأمثلة الواردة في هذا الموضوع فقط, بل أن تبحثوا في الويب -وخاصة في المواقع الأجنبية- عن الفئات وأكوادها وستجدون الكثير من الكنوز التي تستحق عناء البحث والتنقيب, وإن واجهتم أي صعوبة في فهم أحدها, فلا تترددوا في طرحه على بساط البحث والمناقشة, فالعلم صندوق مقفل ومفتاحه السؤال.
تم بحمد الله
اليمن - تعز 18/9/2014

